

Gaetano Marano

Programmi per ATARI 130XE



**GRUPPO EDITORIALE
JACKSON**

DIVISIONE LIBRI

Programmi per ATARI 130XE

Gaetano Marano



GRUPPO
EDITORIALE
JACKSON
Via Rosellini, 12
20124 Milano

© Copyright per l'edizione originale
GRUPPO EDITORIALE JACKSON - Maggio 1986

SUPERVISIONE TECNICA: Mauro Risani
GRAFICA E IMPAGINAZIONE: Moreno Confalone
COPERTINA: Silvana Corbelli
FOTOCOMPOSIZIONE: Jo-Type - Pero (MI)
STAMPA: Stabilimento Grafico A. Matarrelli - Milano

Tutti i diritti sono riservati. Stampato in Italia. Nessuna parte di questo libro può essere riprodotta, memorizzata in sistemi di archivio, o trasmessa in qualsiasi forma o mezzo, elettronico, meccanico, fotocopia, registrazione o altri senza la preventiva autorizzazione scritta dell'editore.

PREFAZIONE

Nell'**Atari 130XE** si cela la più recente versione di un computer almeno altrettanto famoso in America di quanto siano l'Apple II o il Commodore 64.

Il nome è variato negli anni da **Atari 800** (e 400) ad **Atari 800XL** (e 600, 1200, 1450) fino all'attuale **130XE**, così come è cambiato anche l'aspetto esterno lasciando però invariate le caratteristiche per mantenere la compatibilità software con le migliaia di programmi disponibili in America dove gli Atari sono tra i computer più diffusi.

Nel tempo il computer è stato migliorato con l'aggiunta di chip che forniscono dei modi grafici aggiuntivi (progettati da Jay Miner lo stesso che ha progettato i chip grafici di Amiga), di più memoria (128K utilizzabili anche come RAM Disk tramite l'ultima versione di **DOS**, la **2.5**) e di un prezzo sempre più conveniente (oggi il **130XE** è il computer da **128K** con il prezzo più basso).

Nei programmi che descrivo in questo libro, e che sono di tipo molto vario, ho utilizzato le caratteristiche più interessanti del **130XE** come l'alta risoluzione grafica ed il generatore di suoni.

Tra gli altri potrete trovare dei programmi di Business Grafica, un Piano a 3 ottave, dei programmi di disegni e dei programmi per disegnare, dei giochi, eccetera.

Penso quindi che, chi dispone di un 130XE, troverà in questo libro molte cose utili.

Gaetano Marano

USO DEI PROGRAMMI

Tutti i 24 programmi sono registrati sulla cassetta allegata al libro.

Per caricare un programma occorre digitare il comando:

CLOAD

e premere RETURN, quindi, subito dopo il segnale acustico, premere il tasto PLAY del registratore e di nuovo il tasto RETURN del computer; in questo modo viene caricato in memoria il primo programma che si incontra sul nastro.

Tenere presente che a volte, se il nastro non si trova nel punto giusto, il computer può non caricare il programma e dare sullo schermo il messaggio ERROR 13 o 138, in questo caso è sufficiente ripetere le operazioni di caricamento con il nastro nel punto esatto di inizio del programma.

Durante il funzionamento il programma può essere fermato premendo il tasto BREAK, indispensabile anche per uscire dai programmi che non hanno una fine ma che girano in continuazione (oltre a BREAK si può usare anche il tasto RESET).

Elenco dei Programmi

ISTOGRAMMI VERTICALI
ISTOGRAMMI ORIZZONTALI
CILINDRI TRIDIMENSIONALI
DESIGN CARATTERI
SELEZIONE COLORI
PLOTTER
CURVE MATEMATICHE
NEW YORK
DISEGNI DI LINEE
ITALIA
NUMERI DIGITALI
LETTURA VELOCE
CRONOMETRO
SCRITTE PUBBLICITARIE
CONTAPEZZI
FISICA DEI REATTORI NUCLEARI
PIANO 3-OTTAVE
PROGRAMMATORE DI SUONI
SIRENA AMERICANA
CARATTERI IN GRAPHICS 24
TIRO ALLA FUNE
ROULETTE
FLIPPER GIAPPONESE
ROMBOSPIRALE

ISTOGRAMMI VERTICALI

Il programma **Istogrammi Verticali** è il primo di tre programmi di "Business Grafica" vale a dire della rappresentazione in forma grafica di dati economici, tecnici, scientifici, eccetera. Il programma consente di visualizzare, sotto forma di barre verticali, fino ad otto valori numerici contemporaneamente. Facendo girare il programma, il computer chiede per prima cosa di inserire il titolo del grafico, della lunghezza di massimo 80 caratteri.

Quindi chiede l'inserimento degli otto argomenti relativi ai dati e dei corrispondenti dati numerici.

Se si desidera inserire meno di otto dati, ad esempio si vuole un grafico con solo 5 istogrammi, è sufficiente premere RETURN durante l'inserimento dell'argomento del 6° istogramma per far passare il programma alla fase di disegno del grafico.

I valori numerici dei dati inseriti devono essere compresi tra 0 e 5000, penserà poi il programma ad usare la scala più adatta ai valori inseriti; per esempio, se il valore più alto inserito è 897, il programma disegnerà un grafico con un valore massimo di 1000.

Se i valori da inserire sono più grandi di 5000, basta indicare nel titolo del grafico un fattore di moltiplicazione (ad esempio Lire X 1000) ed inserire dei valori numerici divisi per mille.

Dopo la visualizzazione del grafico, è possibile passare all'inserimento dei dati per un nuovo grafico, premendo il tasto ESC.

```

1000 REM * *      Business Grafica      * *
1005 REM * *      ISTOGRAMMI VERTICALI  * *
1010 REM -----
1020 REM      di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1100 POKE 712,0:POKE 710,0:POKE 709,10
1102 POKE 82,0
1105 REM -----
1110 DIM T$(150),R$(8),S$(4)
1120 DIM N$(4),D$(80),A$(40)
1125 FOR G=1 TO 80
1126 D$(G)=" ":T$(G)=CHR$(160)
1127 NEXT G
1130 FOR G=0 TO 7
1132 D$(G*9+1)=CHR$(193+G)
1134 NEXT G
1140 DIM D(8):DIM E(8)
1150 FOR G=1 TO 8
1160 D(G)=0
1170 NEXT G
1210 N=0:PRINT CHR$(125)
1220 PRINT "Inserire titolo grafico (max 80"
1221 PRINT "caratteri) e premere RETURN."
1222 PRINT
1223 INPUT T$
1230 IF LEN(T$)>80 THEN T$=T$(1,80)
1235 T$(LEN(T$)+1)=" "
1240 T$(81)=" "
1250 FOR G=1 TO 8
1260 PRINT CHR$(125)
1270 PRINT "Inserire argomento istogramma ".G;" (max 8"
1271 PRINT "caratteri) e premere RETURN o premere"
1272 PRINT "solo RETURN per meno di 8 dati."
1280 PRINT
1290 INPUT R$
1300 IF R$="" THEN 1400
1310 IF LEN(R$)>8 THEN R$=R$(1,8)
1320 D$((G-1)*9+2)=R$
1325 D$(73)=" "
1330 PRINT
1340 PRINT "Inserire valore ".G;" (0..5000)"
1345 PRINT "e premere RETURN."
1350 PRINT
1360 INPUT Y
1370 IF Y<0 OR Y>5000 THEN 1340
1380 D(G)=Y
1390 NEXT G
1400 GRAPHICS 24:R=0
1405 POKE 712,0:POKE 710,0:POKE 709,10
1410 FOR G=1 TO 8
1420 IF D(G)>R THEN R=D(G)
1430 NEXT G
1440 RESTORE
1450 FOR M=1 TO 9
1460 READ N$,S$
1470 IF VAL(N$)>R THEN 1490
1480 NEXT M
1490 N=VAL(N$)/100
1500 A$="Premere ESC per un'altro grafico."
1510 L=23:C=3:GOSUB 1860
1520 A$=N$:L=8:C=31:GOSUB 1860
1530 A$=S$:L=14:C=31:GOSUB 1860
1540 A$="0":L=20:C=31:GOSUB 1860
1550 L=0:C=0
1560 A$=T$(1,40):GOSUB 1860
1570 A$=T$(41,80):L=1:C=0:GOSUB 1860
1590 FOR G=0 TO 7

```

```

1610 L=3: IF G>3 THEN L=5
1620 IF G=0 OR G=4 THEN C=1
1630 IF G=1 OR G=5 THEN C=11
1640 IF G=2 OR G=6 THEN C=21
1645 IF G=3 OR G=7 THEN C=31
1650 Z=G*9+1
1660 A#=D*(Z,Z+8):GOSUB 1860
1670 NEXT G
1679 COLOR 1
1680 FOR G=65 TO 165 STEP 10
1690 PLOT 37,G:DRAWTO 236,G
1700 IF G=65 OR G=115 OR G=165 THEN PLOT 37,G:DRAWTO 241,G
1710 NEXT G
1712 FOR G=0 TO 7
1713 L=21:C=G*3+6
1714 A#=CHR$(193+G)
1715 GOSUB 1860
1716 NEXT G
1720 PLOT 37,65:DRAWTO 37,165
1730 PLOT 232,65:DRAWTO 232,165
1740 FOR Q=1 TO 8
1750 X=44+(Q-1)*24
1760 V=INT(D(Q)/N*0.5)
1780 IF V>0 THEN GOSUB 1852
1800 NEXT Q
1820 Q=PEEK(764)
1830 IF Q=28 THEN RUN
1850 GOTO 1820
1852 FOR M=X TO (X+15)
1853 PLOT M,164:DRAWTO M,165-V
1854 NEXT M
1855 RETURN
1860 IF A#="" THEN RETURN
1870 SA=PEEK(560)+PEEK(561)*256+282
1880 FOR K=1 TO LEN(A#)
1890 J=ASC(A#(K))
1900 M=0: IF J>127 THEN J=J-128: M=1
1910 IF J>95 THEN GOTO 1940
1920 IF J<32 THEN J=J+64: GOTO 1940
1930 IF J>31 AND J<96 THEN J=J-32
1940 CR=J*8
1950 ST=L*320+C+SA
1960 FOR P=0 TO 7
1970 IF M=0 THEN POKE ST+P*40,PEEK(57344+P+CR)
1980 IF M=1 THEN POKE ST+P*40,255-PEEK(57344+P+CR)
1990 NEXT P
2000 C=C+1
2010 NEXT K
2020 RETURN
2030 DATA 10,5,20,10,50,25,100,50,200,100,500
2040 DATA 250,1000,500,2000,1000,5000,2500

```


ISTOGRAMMI ORIZZONTALI

Il funzionamento del programma **Istogrammi Orizzontali** è del tutto simile a quello del programma precedente tranne per il fatto che i dati vengono visualizzati in maniera diversa (appunto orizzontalmente).

Inoltre, i dati inseriti possono essere compresi tra 0 e 6000 (invece di 5000).

```
1000 REM * *      Business Grafica      * *
1005 REM *  ISTOGRAMMI ORIZZONTALI  *
1010 REM -----
1020 REM -      di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1100 POKE 712,0:POKE 710,208
1102 POKE 709,10:POKE 82,0
1105 REM -----
1110 DIM T$(150),R$(8),S$(4),Q$(4)
1120 DIM H$(4),D$(80),A$(40)
1125 FOR G=1 TO 80
1126 D$(G)="_" : T$(G)=CHR$(150)
1127 NEXT G
1140 DIM D(8):DIM E(8)
1150 FOR G=1 TO 8
1160 D(G)=0
1170 NEXT G
1210 H=0:PRINT CHR$(125)
1220 PRINT "Inserire titolo grafico (max 80"
1221 PRINT "caratteri) e premere RETURN."
1222 PRINT
1223 INPUT T$
1230 IF LEN(T$)>80 THEN T$=T$(1,80)
1235 T$(LEN(T$)+1)=" "
1240 T$(81)=" "
1250 FOR G=1 TO 8
1260 PRINT CHR$(125)
1270 PRINT "Inserire argomento istogramma "G;" (max 8"
1271 PRINT "caratteri) e premere RETURN o premere"
1272 PRINT "solo RETURN per meno di 8 dati."
1280 PRINT
```

```

1290 INPUT R$
1300 IF R$="" THEN 1400
1310 IF LEN(R$)>8 THEN R$=R$(1,8)
1320 D=(G-1)*8+1=R$
1325 D$(65)=" "
1330 PRINT
1340 PRINT "Inserire valore ";G;" (0...6000)"
1345 PRINT "e premere RETURN."
1350 PRINT
1360 INPUT Y
1370 IF Y<0 OR Y>6000 THEN 1340
1380 D(G)=Y
1390 NEXT G
1400 GRAPHICS 24:R=0
1405 POKE 712,0:POKE 710,208:POKE 709,10
1410 FOR G=1 TO 8
1420 IF D(G)>R THEN R=D(G)
1430 NEXT G
1440 RESTORE
1450 FOR M=1 TO 9
1460 READ N$,S$,O$
1470 IF VAL(N$)>=R THEN 1490
1480 NEXT M
1490 N=VAL(N$)/225
1500 A$="Premere ESC per un'altro grafico."
1510 L=23:C=3:GOSUB 1860
1520 A$=N$:L=21:C=38-INT(LEN(N$)/2):GOSUB 1860
1530 A$=S$:L=21:C=29-INT(LEN(S$)/2):GOSUB 1860
1535 A$=O$:L=21:C=19-INT(LEN(O$)/2):GOSUB 1860
1540 A$="0":L=21:C=9:GOSUB 1860
1550 L=0:C=0
1560 A$=T$(1,40):GOSUB 1860
1570 A$=T$(41,80):L=1:C=0:GOSUB 1860
1580 FOR G=1 TO 8
1600 L=G*2+2:C=1
1620 Z=(G-1)*8+1
1625 A$=D$(Z,Z+7)
1630 GOSUB 1860
1670 NEXT G
1679 COLOR 1
1680 FOR G=79 TO 304 STEP 15
1690 PLOT G,24:DRAWTO G,159
1700 IF G=79 OR G=154 OR G=229 OR G=304 THEN PLOT G,28:DRAWTO G,165
1710 NEXT G
1720 PLOT 79,24:DRAWTO 304,24
1730 PLOT 79,155:DRAWTO 304,155
1740 FOR Q=1 TO 8
1750 Y=27+(Q-1)*16
1760 V=INT(D(Q)/N+0,5)
1780 IF V>0 THEN GOSUB 1852
1800 NEXT Q
1820 Q=PEEK(764)
1830 IF Q=28 THEN RUN
1850 GOTO 1820
1852 FOR M=Y TO (Y+12)
1853 PLOT 80,M:DRAWTO 79+V,M
1854 NEXT M
1855 RETURN
1860 IF A$="" THEN RETURN
1870 SA=PEEK(560)+PEEK(561)*256+282
1880 FOR K=1 TO LEN(A$)
1890 J=ASC(A$(K))
1900 M=0 IF J>127 THEN J=J-128:M=1
1910 IF J>95 THEN GOTO 1940
1920 IF J<32 THEN J=J+64:GOTO 1940
1930 IF J>31 AND J<96 THEN J=J-32
1940 CA=J*8
1950 ST=L*320+C+SA
1960 FOR P=0 TO 7
1970 IF M=0 THEN POKE ST+P*40,PEEK(57344+P+CA)
1980 IF M=1 THEN POKE ST+P*40,255-PEEK(57344+P+CA)
1990 NEXT P

```

```
2000 C=C+1
2010 NEXT K
2020 RETURN
2030 DATA 15,10,5,30,20,10,60,40,20
2040 DATA 150,100,50,300,200,100
2050 DATA 500,400,200,1500,1000,500
2060 DATA 3000,2000,1000,6000,4000,2000
```


CILINDRI TRIDIMENSIONALI

L'ultimo programma di Business Grafica visualizza un massimo di sei dati sotto forma di **cilindri tridimensionali**; i dati inseriti possono essere compresi tra 0 e 9999.

```
1000 REM * * Business Grafica * *
1005 REM * CILINDRI TRIDIMENSIONALI *
1010 REM -----
1020 REM di Gaetano Marano -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1100 POKE 712,0:POKE 710,160
1102 POKE 709,10:POKE 82,0
1105 REM -----
1110 DIM T$(150),R$(8),S$(4),D$(4)
1120 DIM N$(4),D$(80),A$(40)
1122 PI=3.14159
1125 FOR G=1 TO 80
1126 D$(G)=" ":T$(G)=CHR$(160)
1127 NEXT G
1140 DIM D(6):DIM E(6)
1150 FOR G=1 TO 6
1160 D(G)=0
1170 NEXT G
1210 N=0:PRINT CHR$(125)
1220 PRINT "Inserire titolo grafico (max 80"
1221 PRINT "caratteri) e premere RETURN."
1222 PRINT
1223 INPUT T$
1230 IF LEN(T$)>80 THEN T$=T$(1:80)
1235 T$(LEN(T$)+1)=" "
1240 T$(81)=" "
1250 FOR G=1 TO 6
1260 PRINT CHR$(125)
1270 PRINT "Inserire argomento cilindro ";G;" (max 5"
1271 PRINT "caratteri) e premere RETURN o premere"
1272 PRINT "solo RETURN per meno di 6 dati."
1280 PRINT
1290 INPUT R$
1300 IF R$="" THEN 1400
1310 IF LEN(R$)>5 THEN R$=R$(1:5)
1320 D$((G-1)*5+1)=R$
1325 D$((31)=" "
1330 PRINT
```

```

1340 PRINT "Inserire valore ";G;" (0_9999)"
1345 PRINT "e premere RETURN."
1350 PRINT
1350 INPUT Y
1370 IF Y<0 OR Y>9999 THEN 1340
1380 D(G)=Y
1390 NEXT G
1400 GRAPHICS 24:R=0
1405 POKE 712,0:POKE 710,160:POKE 709,10
1410 FOR G=1 TO 6
1420 IF D(G)>R THEN R=D(G)
1430 NEXT G
1490 N=R/100
1500 A$="Premere ESC per un'altro grafico."
1510 L=23:C=3:GOSUB 1860
1560 A$=T$(1,40):L=0:C=0:GOSUB 1860
1570 A$=T$(41,80):L=1:C=0:GOSUB 1860
1590 FOR G=1 TO 6
1600 L=21:C=G*5-4
1620 Z=(G-1)*5+1
1625 A$=D$(Z,2+4)
1630 GOSUB 1860
1670 NEXT G
1675 COLOR 1
1676 DEG
1677 PLOT 0,181:DRAWTO 319,181
1680 FOR G=1 TO 6
1690 IF D(G)=0 THEN 1800
1700 F=D(G)/N
1710 X=G*48-12
1720 FOR Q=0 TO 360 STEP 2
1740 KA=19*COS(Q)+X
1750 KB=10*SIN(Q)
1760 PLOT KA,150-KB-F
1770 IF Q=180 THEN PLOT KA,150-KB:DRAWTO KA,150-KB-F
1780 NEXT Q
1790 C=G*5-4:L=INT(18-(D(G)/N+20)/8)
1792 A$=STR$(INT(D(G)))
1795 GOSUB 1860
1800 NEXT G
1820 Q=PEEK(764)
1830 IF Q=28 THEN RUN
1850 GOTO 1820
1860 IF A$="" THEN RETURN
1870 SA=PEEK(560)+PEEK(561)*256+282
1880 FOR K=1 TO LEN(A$)
1890 J=ASC(A$(K))
1900 M=0:IF J>127 THEN J=J-128:M=1
1910 IF J>95 THEN GOTO 1940
1920 IF J<32 THEN J=J+64:GOTO 1940
1930 IF J>31 AND J<96 THEN J=J-32
1940 CA=J*8
1950 ST=L*320+C+SA
1960 FOR P=0 TO 7
1970 IF M=0 THEN POKE ST+P*40,PEEK(57344+P+CA)
1980 IF M=1 THEN POKE ST+P*40,255-PEEK(57344+P+CA)
1990 NEXT P
2000 C=C+1
2010 NEXT K
2020 RETURN

```

DESIGN CARATTERI

Il set di caratteri dell'**Atari 130XE** può essere ridefinito riscrivendo i codici dalla ROM alla RAM e modificandoli con un breve programma nel modo descritto più avanti.

Il programma "**Design Caratteri**" serve per disegnare i caratteri (che hanno una matrice di 8 x 8 punti) e calcolare in modo automatico i codici da scrivere nella RAM per visualizzare i nuovi caratteri o simboli definiti.

Il disegno del nuovo carattere si può realizzare usando i quattro tasti di controllo del cursore (da premere senza il tasto Control), oltre a questi si può usare il tasto M per spostarsi senza disegnare o per fare delle modifiche, il tasto D per tornare al modo disegno, il tasto X per avere i codici del carattere disegnato fino a quel momento, il tasto C per avere la copia su stampante del carattere e dei codici ed il tasto ESC che annulla tutto e fa ripartire il programma.

Il tasto C che produce la copia su carta va premuto solo se la stampante è collegata ed in funzione altrimenti il computer dà un messaggio di errore (ERROR 138) ed il programma si ferma.

L'uso dei codici ottenuti può essere descritto con un breve programma.

L'indirizzo della ROM a partire dal quale si trovano i codici dei caratteri standard (57344 all'accensione del computer) è contenuto nella variabile di sistema alla locazione di memoria 756 (con il numero 224 valore che va può essere letto con una PEEK e che va moltiplicato per 256).

Prima però di modificare tale valore occorre duplicare i codici dei caratteri dalla ROM alla RAM, e questo specialmente se solo parte dei caratteri deve essere modificata. Le istruzioni per trasferire i codici sono:

```

100 FOR A = 38912 TO 39935
110 POKE A PEEK (A+18432)
120 NEXT A

```

I codici sono in tutto 1024 (8 byte per ogni carattere) relativi ai primi 128 caratteri (da CHR\$ 0 a CHR\$ 127), gli altri caratteri (da CHR\$ 128 a CHR\$ 255) vengono prodotti dal computer per inversione.

L'area della RAM scelta per memorizzare i caratteri (da 38192 a 39935) è libera se il computer è nei modi GRAPHICS 0, 1 o 2, tutti modi che usano meno di 1K di RAM (in tali "modi" l'area usata dal processore video è tra 39936 e 40959).

Se si usano nel programma anche altri modi grafici i caratteri vanno memorizzati più in basso (ad esempio nel modo GRAPHICS 8 o 24 la locazione iniziale deve essere altri 7K più in basso, all'indirizzo 31744).

Memorizzati i caratteri nella RAM si può modificare con una POKE la variabile di sistema che indica al computer dove cercarli:

```

130 POKE 756 152

```

Il valore 152 si ricava da 38192/256. Spostati i caratteri nella RAM è ora possibile modificarli tutti o solo alcuni sostituendo con una breve routine i codici originali con i codici ricavati usando il programma per disegnare i caratteri, codici contenuti in una serie di istruzioni DATA.

Nel nostro esempio vengono modificati solo tre caratteri (CHR\$ 33, 34, e 35):

```

140 FORWA = 38912+8 TO 3991+31
150 READ D: POKE A, D
160 NEXT A
170 DATA 255, 129, 129, 129, 129, 129, 129, 255
180 DATA 255, 129, 189, 165, 165, 189, 129, 255
190 DATA 56, 56, 16, 124, 146, 16, 56, 108

```

Da notare che i 128 caratteri nella ROM (e quindi adesso anche nella RAM) non si trovano in ordine ma sono sistemati in una maniera particolare, infatti vengono prima i 64 caratteri da CHR\$ 32 a CHR\$ 95, quindi i 32 caratteri da CHR\$ 0 a CHR\$ 31 ed infine i 32 caratteri da CHR\$ 96 a CHR\$ 127. Inoltre occorre ricordare che ogni carattere occupa 8 byte così, ad esempio, il carattere "a" minuscola (CHR\$ 97) inizia nella RAM alla locazione 39688 ($97 \times 8 + 38912$) e finisce alla locazione 39695 ($98 \times 8 + 38912 - 1$).

Modificati i tre caratteri nel modo appena descritto è ora possibile visualizzarli con qualche semplice istruzione:

```
200 FOR A = 1 TO 8
210 PRINT, CHR$(33), CHR$(34), CHR$(35)
220 PRINT
230 NEXT A
240 END
```

Con i nuovi caratteri nella RAM è anche possibile passare da un set all'altro durante il programma semplicemente cambiando con una POKE il valore della locazione 756 con 224 o 152.

```
1000 REM * *   DESIGN CARATTERI   * *
1010 REM -----
1020 REM      di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0: SOUND 1,0,0,0
1075 SOUND 2,0,0,0: SOUND 3,0,0,0
1090 POKE 710,0: POKE 712,98: POKE 709,10
1100 POKE 82,0: POKE 752,1
1110 REM -----
1120 DIM M(8,8), T(8), P$(8)
1122 FOR A=1 TO 8
1123 T(A)=0
1124 FOR B=1 TO 8
1125 M(A,B)=0
1126 NEXT B
1127 NEXT A
1130 DIM A$(10), B$(10)
1140 FOR A=1 TO 10
1150 A$(A)=CHR$(18)
1160 B$(A)=CHR$(18)
1170 NEXT A
1180 A$(1)=CHR$(17)
1190 A$(10)=CHR$(5)
```

```

1200 B$(1)=CHR$(26)
1210 B$(10)=CHR$(3)
1220 POSITION 13,1:PRINT A$
1230 FOR A=2 TO 9
1240 POSITION 13,A:PRINT "I-----I"
1250 NEXT A
1260 POSITION 13,10:PRINT B$
1270 POSITION 5,14:PRINT "Cursori -- Tasti per disegnare"
1280 POSITION 5,16:PRINT "D -- Disegno X -- Dati finali"
1290 POSITION 5,18:PRINT "M ---- Modifiche e spostamenti"
1300 POSITION 5,20:PRINT "C -- Copy ESC -- Nuovo caratt."
1310 X=14:Y=2:D=1:M(1,1)=1
1320 GOSUB 1570
1330 K=PEEK(764)
1340 POKE 764,0
1350 IF K=6 THEN GOSUB 1470
1360 IF K=7 THEN GOSUB 1490
1370 IF K=14 THEN GOSUB 1510
1380 IF K=15 THEN GOSUB 1530
1390 IF K=58 THEN GOSUB 1550
1400 IF K=37 THEN GOSUB 1560
1410 IF K=22 THEN GOSUB 1570
1420 IF K=18 THEN GOSUB 1670
1430 IF K=28 THEN RUN
1440 POSITION X,Y:PRINT CHR$(160)
1450 IF D=0 THEN POSITION X,Y:PRINT "--"
1455 M(X-13,Y-1)=D
1460 GOTO 1330
1470 IF X=14 THEN RETURN
1480 X=X-1:RETURN
1490 IF X=21 THEN RETURN
1500 X=X+1:RETURN
1510 IF Y=2 THEN RETURN
1520 Y=Y-1:RETURN
1530 IF Y=9 THEN RETURN
1540 Y=Y+1:RETURN
1550 D=1:RETURN
1560 D=0:RETURN
1570 FOR A=1 TO 8
1580 T(A)=0
1590 FOR B=7 TO 0 STEP -1
1600 V=2^B
1610 T(A)=T(A)+V*(B-B.A)
1620 NEXT B
1630 POSITION 24,A+1:PRINT "      <"
1635 FOR Z=1 TO 50:NEXT Z
1640 POSITION 24,A+1:PRINT T(A);"      "
1650 NEXT A
1660 RETURN
1670 GOSUB 1570:LPRINT " "
1680 FOR A=1 TO 8
1685 P$=" "
1690 FOR B=1 TO 8
1700 IF M(B,A)=1 THEN P$(B)="*"
1710 IF M(B,A)=0 THEN P$(B)="-"
1720 NEXT B
1730 LPRINT P$;" " ;T(A)
1740 NEXT A
1750 LPRINT " " :LPRINT " " :LPRINT " "
1760 RETURN

```

SELEZIONE COLORI

Un'altra caratteristica interessante del 130XE è quella di disporre di 128 diverse sfumature di colore, anche se non tutte visualizzabili contemporaneamente nei vari modi grafici, testo e testo/grafica.

Comunque per rendere più veloce la scelta dei colori (che sono 16) e delle intensità (che sono 8) può essere usato questo programma che consente di vedere facilmente sullo schermo le varie sfumature.

Il programma disegna prima una serie di quadrati, il loro colore può essere quindi scelto premendo uno dei tasti 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, A, B, C, D, E, F, la luminosità invece può essere aumentata o diminuita premendo i due tasti di controllo cursore freccia in alto e freccia in basso (senza premere insieme il tasto Control).

Nella finestra testo in basso vengono sempre visualizzati i codici da inserire nella istruzione SETCOLOR per ottenere lo stesso colore presente sullo schermo.

```
1000 REM * *   SELEZIONE COLORI   * *
1010 REM -----
1020 REM -       di Gaetano Marano       -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 3
1070 SOUND 0:0.0,0:0:SOUND 1:0.0,0
1075 SOUND 2:0.0,0:0:SOUND 3:0.0,0
1090 POKE 710:0:POKE 712:0:POKE 709,10
1100 POKE 82:0:POKE 752,1:POKE 708,118
1110 REM -----
1115 COLOR 1
1120 FOR A=1 TO 400
1130 PLOT RND(0)*39,RND(0)*19
1140 NEXT A
1145 C=7 I=6:GOSUB 1400
1150 K=PEEK(764)
```

```

1160 POKE 764,0
1170 IF K=50 THEN C=0:GOSUB 1400
1180 IF K=31 THEN C=1:GOSUB 1400
1190 IF K=30 THEN C=2:GOSUB 1400
1200 IF K=26 THEN C=3:GOSUB 1400
1210 IF K=24 THEN C=4:GOSUB 1400
1220 IF K=29 THEN C=5:GOSUB 1400
1230 IF K=27 THEN C=6:GOSUB 1400
1240 IF K=51 THEN C=7:GOSUB 1400
1250 IF K=53 THEN C=8:GOSUB 1400
1260 IF K=48 THEN C=9:GOSUB 1400
1270 IF K=53 THEN C=10:GOSUB 1400
1280 IF K=21 THEN C=11:GOSUB 1400
1290 IF K=18 THEN C=12:GOSUB 1400
1300 IF K=58 THEN C=13:GOSUB 1400
1310 IF K=42 THEN C=14:GOSUB 1400
1320 IF K=56 THEN C=15:GOSUB 1400
1330 IF K=14 THEN GOSUB 1360:GOSUB 1400
1340 IF K=15 THEN GOSUB 1380:GOSUB 1400
1350 GOTO 1150
1360 IF I=14 THEN RETURN
1370 I=I+2:RETURN
1380 IF I=0 THEN RETURN
1390 I=I-2:RETURN
1400 SETCOLOR 0,C,I
1410 PRINT CHR$(125)
1420 PRINT
1430 PRINT "      COLORE __ ";C;"      INTENSITA' __ ";I
1440 RETURN

```


PLOTTER

Plotter è un programma per disegnare nel modo ad alta risoluzione (320 x 192 pixel).

I tasti per disegnare sono otto (dal tasto 3 al tasto 0) come indicato nella tabella in fondo.

Per modificare il disegno o spostarsi si deve premere il tasto 2 mentre per ritornare al modo disegno va premuto il tasto 1.

Tasto	Funzione
1	modo "disegno"
2	modo "modifiche e spostamenti"
3	in alto a sinistra
4	in basso a sinistra
5	a sinistra
6	in basso
7	in alto
8	a destra
9	in basso a destra
0	in alto a destra

```
1000 REM '* * *      PLOTTER      * * *
1010 REM -----
1020 REM -      di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 24
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1080 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 712,112:POKE 710,0:POKE 709,14
1100 POKE 62,0:POKE 752,1
1110 REM -----
1350 X=50:Y=50:D=1
1355 COLOR 1
1380 K=PEEK(764)
1390 POKE 764,0
1400 IF K=51 THEN GOSUB 1680
1405 IF K=27 THEN GOSUB 1700
```

```

1410 IF K=29 THEN GOSUB 1660
1415 IF K=53 THEN GOSUB 1720
1420 IF K=26 THEN GOSUB 1640
1425 IF K=50 THEN GOSUB 1740
1430 IF K=24 THEN GOSUB 1620
1435 IF K=48 THEN GOSUB 1580
1438 IF K=31 THEN GOSUB 1600
1442 IF K=30 THEN GOSUB 1610
1450 COLOR 1
1460 PLOT X,Y
1470 COLOR 0
1480 PLOT X,Y
1570 K=0:GOTO 1380
1580 IF X=319 OR Y=191 THEN RETURN
1590 X=X+1:Y=Y+1:RETURN
1600 D=1:RETURN
1610 D=0:RETURN
1620 IF X=0 OR Y=191 THEN RETURN
1630 X=X-1:Y=Y+1:RETURN
1640 IF X=0 OR Y=0 THEN RETURN
1650 X=X-1:Y=Y-1:RETURN
1660 IF X=0 THEN RETURN
1670 X=X-1:RETURN
1680 IF Y=0 THEN RETURN
1690 Y=Y-1:RETURN
1700 IF Y=191 THEN RETURN
1710 Y=Y+1:RETURN
1720 IF X=319 THEN RETURN
1730 X=X+1:RETURN
1740 IF X=319 OR Y=0 THEN RETURN
1750 X=X+1:Y=Y-1:RETURN

```

CURVE MATEMATICHE

Questo programma disegna in alta risoluzione le **curve** di alcune funzioni **matematiche**.

```
1000 REM * * CURVE MATEMATICHE * *
1010 REM -----
1020 REM - di Gaetano Marano -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 24
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 712,126:POKE 710,120:POKE 709,112
1095 POKE 82,0
1098 REM -----
1100 PI=3.14159
1120 COLOR 1
1140 FOR X=0 TO 318
1150 PLOT X,191-(XX/530)
1160 NEXT X
1170 FOR X=0 TO 318
1180 PLOT X,191-(XX/530):DRAWTO X,191
1190 NEXT X
1200 FOR A=1 TO 2000:NEXT A
1210 GRAPHICS 24
1212 POKE 712,206:POKE 710,200:POKE 709,192
1220 FOR X=0 TO 318
1230 Y=RND(0)*X/1.7
1240 PLOT X,190-Y:DRAWTO X,190
1250 NEXT X
1260 FOR A=1 TO 2000:NEXT A
1270 GRAPHICS 24
1272 POKE 712,62:POKE 710,56:POKE 709,48
1280 FOR X=0 TO 318
1290 Y=191-LOG(X/5+1)*45
1300 PLOT X,Y
1310 NEXT X
1320 FOR X=0 TO 318
1330 Y=191-LOG(X/5+1)*45
1340 PLOT X,Y:DRAWTO X,191
1350 NEXT X
1360 FOR A=1 TO 2000:NEXT A
1365 GRAPHICS 24
1367 POKE 712,222:POKE 710,216:POKE 709,208
1380 FOR X=0 TO 318
1390 PLOT X,191-(95+COS(PI/180*(180+X*1.13)))*95)
1400 NEXT X
1410 FOR X=0 TO 318
1420 Y=95+COS(PI/180*(180+X*1.13))*95)
1430 PLOT X,191-Y:DRAWTO X,191
1440 NEXT X
1450 FOR A=1 TO 3000:NEXT A
1460 GRAPHICS 0
```


NEW YORK

Il programma **New York** disegna sullo schermo il panorama di una città con due file di grattacieli.

Ogni volta che si fa girare, il programma crea un panorama diverso e casuale.

```
1000 REM * * *      NEW YORK      * * *
1010 REM -----
1020 REM -      di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 11
1070 SOUND 0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1080 POKE 82,0
1110 REM -----
1130 FOR A=1 TO 20
1140 COLOR 2+INT(RND(0)*14)
1150 PLOT RND(0)*79,RND(0)*80
1160 NEXT A
1170 K=5
1180 DIM L(15)
1190 DIM H(15)
1200 DIM I(15)
1210 FOR A=1 TO 15
1220 L(A)=((2+INT(RND(0)*3))*2)
1230 H(A)=(K+INT(RND(0)*(2+K)))*8+INT((RND(0)*K)-1)
1240 I(A)=2+INT(RND(0)*14)
1250 NEXT A
1260 X=0
1270 FOR A=1 TO 15
1280 X=X+L(A)
1290 IF X<80 THEN GOSUB 1400
1300 IF X>79 THEN GOTO 1320
1310 NEXT A
1320 L(A)=79-(X-L(A))
1330 X=79
1340 GOSUB 1400
1350 IF K=2 THEN 1380
1360 K=2
1370 GOTO 1210
1380 COLOR 1
1390 GOTO 1390
1400 SETCOLOR 4,0,8:COLOR I(A)
1410 FOR Y=0 TO H(A) STEP 2
1420 PLOT X-L(A),191-Y:DRAWTO X-1,191-Y
1430 COLOR 0
1440 PLOT X-L(A),190-Y:DRAWTO X-1,190-Y
1450 COLOR I(A)
1460 NEXT Y
1470 PLOT X-L(A),191:DRAWTO X-L(A),192-H(A)
1480 PLOT X-1,191:DRAWTO X-1,192-H(A)
1490 RETURN
```


DISEGNI DI LINEE

Il programma disegna sullo schermo in alta risoluzione quattro diverse immagini grafiche una di seguito all'altra. In particolare il terzo disegno può anche essere variato cambiando il valore della variabile V alla linea 1360 con un valore diverso, sia maggiore che minore di 12%

```
1000 REM * *      DISEGNI DI LINEE      * *
1010 REM -----
1020 REM -      di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 24
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1080 SOUND 2,0,0,0:SOUND 3,0,0,0
1100 SETCOLOR 4,0,0:SETCOLOR 2,0,0:SETCOLOR 1,0,10
1105 POKE 82,0
1106 REM -----
1110 COLOR 1
1120 PI=3.14159
1140 FOR A=0 TO 360 STEP 2
1150 K=5+15*RNDRND(0)
1160 RD=PI/180*K*A
1170 H=40+50*RNDRND(0)
1180 PLOT 160+K*COS(RD),96+K*SIN(RD):DRAWTO 160+H*COS(RD),96+H*SIN(RD)
1190 NEXT A
1200 FOR T=1 TO 2000:NEXT T
1205 GRAPHICS 24
1210 SETCOLOR 2,0,0
1220 FOR A=7 TO 95 STEP 8
1230 PLOT 159,A+96:DRAWTO 57+A,96
1240 NEXT A
1250 FOR A=7 TO 95 STEP 8
1260 PLOT 160,A+96:DRAWTO 262-A,96
1270 NEXT A
1280 FOR A=0 TO 95 STEP 8
1290 PLOT 159,A:DRAWTO 152-A,95
1300 NEXT A
1310 FOR A=0 TO 95 STEP 8
1320 PLOT 160,A:DRAWTO 167+A,95
1330 NEXT A
1340 FOR T=1 TO 2000:NEXT T
1345 GRAPHICS 24
1350 SETCOLOR 2,0,0
1360 V=12
1370 DIM X(V)
1380 DIM Y(V)
1390 PI=PI*2
1400 FOR N=1 TO V
```

```

1410 A=PI/V*N
1420 X(N)=COS(A)*88+160
1430 Y(N)=SIN(A)*95+95
1440 NEXT N
1450 FOR A=1 TO V
1460 FOR B=1 TO V
1470 PLOT X(A),Y(A):DRAWTO X(B),Y(B)
1480 NEXT B
1490 NEXT A
1500 FOR T=1 TO 2000:NEXT T
1505 GRAPHICS 24
1510 SETCOLOR 2,0,0
1515 PI=PI/2
1520 I=100:A=0:B=A+I:N=1
1530 R=A*PI/180
1540 XA=95*COS(R)+160
1550 YA=95*SIN(R)+95
1560 R=B*PI/180
1570 XB=95*COS(R)+160
1580 YB=95*SIN(R)+95
1590 PLOT XA,YA:DRAWTO XB,YB
1600 A=A+I:B=A+I:N=N+1
1610 IF N=19 THEN 1630
1620 GOTO 1530
1630 FOR T=1 TO 3000:NEXT T

```


ITALIA

La serie di numeri inseriti nelle istruzioni **DATA** di questo programma, contengono le coordinate per disegnare i contorni dell'Italia.

Così com'è il programma è più che altro dimostrativo, tuttavia la parte principale che disegna l'Italia può essere inserita in altri programmi di tipo didattico, di quiz, eccetera.

Il punto di inizio del disegno, dipende dai valori assegnati alle variabili X e Y (linee 1120 e 1130).

```
1000 REM * * *          ITALIA          * * *
1010 REM -----
1020 REM      di Gaetano Marano      -
1030 REM -----
1040 REM <c> Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 24
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1078 SETCOLOR 4,0,0:SETCOLOR 2,0,0
1079 SETCOLOR 1,0,14
1080 POKE 82,0
1110 REM -----
1115 COLOR 1
1120 X=99:REM 0...224
1130 Y=30:REM 0...76
1140 RESTORE 1260
1160 READ D
1170 IF D=1111 THEN 1250
1180 IF D=0 THEN Y=Y+1:GOTO 1230
1200 D=D-52
1220 PLOT 10+X+D,Y
1230 GOTO 1160
1250 GOTO 1250
1260 DATA 94,0,88,89,91,92,93,0,86,87,88,90,93,0,82,83,85,94,0,81,83,84,95,0
1270 DATA 79,82,96,97,98,99,0,78,80,81,100,101,102,103,0,64,65,72,78,104,105,0
1280 DATA 64,65,71,73,74,75,76,77,78,103,0,64,66,71,103,104,0
1290 DATA 63,67,68,70,105,0,59,61,62,68,70,104,0,54,55,56,57,58,60,69,104,0
1300 DATA 53,101,104,0,54,100,101,102,103,104,105,0,55,100,101,106,107,0
1310 DATA 55,99,107,0,56,95,96,97,98,0,55,94,0,54,94,0
1320 DATA 52,53,94,0,52,95,0,53,95,0,53,96,0,54,94,95,96,0,53,94,0
1330 DATA 53,94,0,53,94,0,53,67,68,69,94,0,53,65,66,70,71,94,0
1340 DATA 54,55,64,72,73,95,0,56,57,58,59,63,73,74,95,0,59,62,75,76,96,0
1350 DATA 59,62,77,97,0,58,59,60,61,78,98,0,78,99,100,0,79,101,0,79,102,0
1360 DATA 79,103,0,79,104,105,0,80,105,0,80,105,0,81,106,0,81,106,0,75,81,107,0
1370 DATA 81,82,107,0,83,107,0,77,78,79,83,107,0,79,84,108,0,76,85,108,0
1380 DATA 86,109,0,86,87,88,110,0,83,85,89,111,0,90,112,0,90,113,114,0
1390 DATA 91,92,115,0,93,116,117,0,94,118,119,122,123,124,125,0,94,120,121,126,0
1400 DATA 95,125,0,96,124,0,97,124,0,98,99,125,126,0,69,100,102,103,127,128,129
```

```

1410 DATA 0,62,69,101,104,105,106,107,130,0,61,67,68,70,71,108,131,132,133,0
1420 DATA 96,72,108,134,135,0,60,65,71,109,136,0
1430 DATA 60,61,62,63,64,72,109,110,111,137,138,0,60,72,109,112,139,140,0
1440 DATA 60,61,72,107,113,114,115,141,142,0,62,73,112,113,116,134,135,143,0
1450 DATA 62,73,110,117,133,136,144,0,62,72,117,132,137,138,139,140,145,0
1460 DATA 62,72,117,132,142,146,0,63,71,117,118,131,143,147,0
1470 DATA 62,72,119,122,131,143,146,0,62,72,120,121,123,130,143,144,146,0
1480 DATA 62,72,123,124,131,145,146,0,63,72,124,130,0,63,72,124,130,0
1490 DATA 62,63,72,125,130,0,62,71,125,131,132,133,0,61,71,126,134,0
1500 DATA 62,71,126,135,0,61,71,127,135,0,59,61,67,68,69,71,127,135,0
1510 DATA 62,66,70,127,135,0,61,63,66,127,135,0,63,64,65,128,133,134,135,0
1520 DATA 128,132,0,128,131,0,125,126,127,131,0,118,126,131,0,116,126,131,0
1530 DATA 126,130,131,0,118,125,129,0,122,124,125,128,0
1540 DATA 103,117,120,121,122,124,128,0,98,101,102,103,114,115,116,118,119
1550 DATA 122,124,128,0,92,97,99,100,104,105,109,110,111,112,113,121,124,125
1560 DATA 126,127,0,94,96,106,107,108,120,0,96,120,0,96,119,0,97,119,0
1570 DATA 98,99,100,101,119,0,102,103,118,0,104,118,0,105,106,118,0
1580 DATA 107,108,120,0,109,110,111,112,121,0,113,120,0,114,119,0
1590 DATA 91,114,115,118,119,0,92,116,117,0,1111

```

NUMERI DIGITALI

Se si vogliono inserire dei numeri di varie dimensioni nel modo grafico ad alta risoluzione, si può usare la routine Numeri Digitali che consente appunto di scrivere delle cifre digitali a sette segmenti.

Per scrivere un numero bisogna inserire il numero stesso (tra 0 e 9) nella variabile N, la larghezza dei segmenti orizzontali (in pixel) nella variabile L, l'altezza dei segmenti verticali nella variabile H e le coordinate alle quali va scritto va eseguita un'istruzione GOSUB 7000.

Se le cifre devono essere tutte della stessa dimensione, i valori L e H possono essere stabiliti una sola volta all'inizio, inoltre se si devono scrivere più numeri di seguito, occorre ad ogni numero incrementare il valore di X per non sovrapporre le cifre.

```
100 REM * Programma dimostrativo *
102 REM -----
104 GRAPHICS 24
107 POKE 712,114:POKE 710,0:POKE 709,10
108 POKE 82,0
116 FOR A=1 TO 20
118 X=RND(1)*257
120 Y=RND(1)*127
122 L=2+RND(1)*30
124 H=2+RND(1)*30
126 N=RND(1)*10
128 GOSUB 7000
130 NEXT A
132 FOR T=1 TO 3000:NEXT T
134 END
7000 REM * * * NUMERI DIGITALI * * *
7010 REM -----
7020 REM - di Gaetano Marano -
7030 REM -----
7040 REM (c) Gruppo Editoriale Jackson
7050 REM -----
7055 COLOR 1
7060 N=INT(N):IF N<0 OR N>9 THEN RETURN
7070 IF N=0 OR N=2 OR N=3 OR N=5 OR N=6 OR N=7 OR N=8 OR N=9 THEN PLOT X+32,Y+4:
DRAWTO X+32+L,Y+4
7080 IF N=2 OR N=3 OR N=4 OR N=5 OR N=6 OR N=8 OR N=9 THEN PLOT X+32,Y+4+H:DRAWTO
X+32+L,Y+4+H
```

```

7090 IF N=0 OR N=2 OR N=3 OR N=5 OR N=6 OR N=8 OR N=9 THEN PLOT X+32,Y+4+H+H:DRA
WTO X+32+L,Y+4+H+H
7100 IF N=0 OR N=1 OR N=2 OR N=3 OR N=4 OR N=7 OR N=8 OR N=9 THEN PLOT X+32+L,Y+
4:DRAWTO X+32+L,Y+4+H
7110 IF N=0 OR N=1 OR N=3 OR N=4 OR N=5 OR N=6 OR N=7 OR N=8 OR N=9 THEN PLOT X+
32+L,Y+4+H:DRAWTO X+32+L,Y+4+H+H
7120 IF N=0 OR N=2 OR N=6 OR N=8 THEN PLOT X+32,Y+4+H:DRAWTO X+32,Y+4+H+H
7130 IF N=0 OR N=4 OR N=5 OR N=6 OR N=8 OR N=9 THEN PLOT X+32,Y+4:DRAWTO X+32,Y+
4+H
7140 RETURN

```

LETTURA VELOCE

Questo programma può essere usato per incrementare la propria velocità di lettura. In una serie di istruzioni DATA sono contenute 200 parole che vengono visualizzate in maniera casuale e per breve tempo in un riquadro sullo schermo.

La prima frase viene sempre visualizzata per un tempo di 0.1 secondi che viene indicato in anticipo.

Sparita la frase il computer chiede di scriverla, quindi confronta la frase scritta con quella appena visualizzata.

Se la frase inserita è uguale a quella scelta dal computer esce la scritta "Frase: ESATTA", se invece la frase inserita è diversa da quella visualizzata esce la scritta

Frase: SBAGLIATA!!

Inoltre, ogni volta che si riconosce esattamente una frase, il tempo di visualizzazione della successiva viene dimezzato (fino ad un minimo di 0.02 secondi), mentre ogni volta che si sbaglia viene raddoppiato (fino ad un massimo di 1 secondo.)

Alle 200 parole già incluse nel programma se ne possono aggiungere molte altre, inserendo le nuove parole o frasi (di massimo 20 caratteri) in una serie di istruzioni DATA; l'ultima linea deve essere sempre uguale alla linea 998 (vedere lista-to) che contiene un carattere usato dal programma per riconoscere la fine della lista durante il conteggio delle frasi.

```

1000 REM * * * LETTURA VELOCE * * *
1010 REM -----
1020 REM - di Gaetano Marano
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0: SOUND 1,0,0
1080 SOUND 2,0,0: SOUND 3,0,0
1090 POKE 710,0: POKE 712,0: POKE 709,10
1100 POKE 82,0: POKE 752,1
1105 REM -----
1110 DIM T$(12),B$(80),C$(20),F$(20),X$(20)
1115 DIM A$(1),D$(1),E$(1),G$(1),H$(20),I$(2)
1118 A$=CHR$(17):D$=CHR$(26):E$=CHR$(5):G$=CHR$(3):H$=CHR$(18):I$=CHR$(124)
1120 B$=""
"
1150 POSITION 9,5:PRINT A$;H$;H$;H$;H$;H$;H$;H$;H$;H$
1155 POSITION 20,5:PRINT H$;H$;H$;H$;H$;H$;H$;H$;H$;E$
1160 POSITION 9,6:PRINT I$: POSITION 30,6:PRINT I$
1180 POSITION 9,7:PRINT D$;H$;H$;H$;H$;H$;H$;H$;H$;H$;H$;H$;H$
1190 POSITION 20,7:PRINT H$;H$;H$;H$;H$;H$;H$;H$;H$;G$
1200 RESTORE
1210 FOR A=1 TO 10000
1220 READ C$: IF C$="" THEN 1240
1230 NEXT A
1240 NF=A-1
1245 TX=3
1250 IF TX=1 THEN T$="0040,02 Sec."
1251 IF TX=2 THEN T$="0100,05 Sec."
1252 IF TX=3 THEN T$="0200,1 Sec."
1253 IF TX=4 THEN T$="0400,2 Sec."
1254 IF TX=5 THEN T$="1000,5 Sec."
1255 IF TX=6 THEN T$="200 1 Sec."
1330 FOR T=1 TO 800:NEXT T
1350 POSITION 9,10:PRINT "Durata Frase ";T$(4,LEN(T$))
1360 FOR T=1 TO 400:NEXT T
1370 G=1+INT(NF*RND(0))
1380 RESTORE G+2000
1390 READ F$
1400 P=VAL(T$(1,3))
1410 IF LEN(F$)>20 THEN F$=F$(1,20)
1420 POSITION 10,6:PRINT F$
1430 FOR T=1 TO P:NEXT T
1440 POSITION 10,6:PRINT " "
1450 FOR T=1 TO 240:NEXT T
1460 POSITION 4,10:PRINT "Scrivete la frase che avete visto : e premete RETU
RN"
1470 POSITION 4,12:INPUT X$
1480 R=0: IF X$=F$ THEN R=1
1490 IF R=1 THEN POSITION 4,10:PRINT " Frase: ** ESATTA **";B$;B$
1500 IF R=0 THEN POSITION 4,10:PRINT " Frase: SBAGLIATA !!";B$;B$
1510 IF R=1 AND TX>1 THEN TX=TX-1
1520 IF R=0 AND TX<5 THEN TX=TX+1
1530 GOTO 1250
2000 REM ---- CATALOGO PAROLE E FRASI
2001 DATA BIANCO
2002 DATA ROSSO
2003 DATA VERDE
2004 DATA GIALLO
2005 DATA BLU
2006 DATA NERO
2007 DATA ROMA
2008 DATA MILANO
2009 DATA FIRENZE
2010 DATA NAPOLI
2011 DATA PARIGI
2012 DATA BERLINO
2013 DATA LONDRA
2014 DATA MADRID
2015 DATA NEW YORK
2016 DATA LOS ANGELES
2017 DATA ATENE

```

2018 DATA TOKYO
 2019 DATA MOSCA
 2020 DATA BUENOS AIRES
 2021 DATA GINEVRA
 2022 DATA MICROPROCESSORE
 2023 DATA KEYBOARD
 2024 DATA MONITOR
 2025 DATA ATARI 130XE
 2026 DATA AEREO
 2027 DATA NAVE
 2028 DATA TRENO
 2029 DATA COMPUTER
 2030 DATA FOTOGRAFIA
 2031 DATA ATTRICE
 2032 DATA AUTOMOBILE
 2033 DATA CREDIT CARD
 2034 DATA LINEA
 2035 DATA ARTICOLO
 2036 DATA INDIRIZZO
 2037 DATA MOTORE
 2038 DATA CAPIRE
 2039 DATA VOLARE
 2040 DATA CAMMINARE
 2041 DATA NAPOLEONE
 2042 DATA GARIBALDI
 2043 DATA CESARE
 2044 DATA WASHINGTON
 2045 DATA PRINTER
 2046 DATA UOMO
 2047 DATA DONNA
 2048 DATA CASA
 2049 DATA SCALA
 2050 DATA ASCENSORE
 2051 DATA RUOTA
 2052 DATA LIBRO
 2053 DATA JACKSON
 2054 DATA BYTE
 2055 DATA BUONO
 2056 DATA CATTIVO
 2057 DATA VELOCE
 2058 DATA LENTO
 2059 DATA GRANDE
 2060 DATA PICCOLO
 2061 DATA LEGGERO
 2062 DATA PESANTE
 2063 DATA MARIO
 2064 DATA LAURA
 2065 DATA FRANCO
 2066 DATA ANTONIO
 2067 DATA MARIA
 2068 DATA FRANCESCO
 2069 DATA BARBARA
 2070 DATA GATTI
 2071 DATA ELEFANTE
 2072 DATA DELFINO
 2073 DATA GIRAFFA
 2074 DATA ANTILOPE
 2075 DATA LEONE
 2076 DATA CONDOR
 2077 DATA TIGRE
 2078 DATA DOBERMAN
 2079 DATA IPPOPOTAMO
 2080 DATA CANE
 2081 DATA BASIC
 2082 DATA PASCAL
 2083 DATA FORTRAN
 2084 DATA FORTH
 2085 DATA LISP
 2086 DATA PROLOG
 2087 DATA LOGO
 2088 DATA COBOL
 2089 DATA APL
 2090 DATA CALENDARIO

2091 DATA TRANSISTOR
2092 DATA STOFFA
2093 DATA TEATRO
2094 DATA RIVISTA
2095 DATA SEMAFORO
2096 DATA VOLERE
2097 DATA CANTARE
2098 DATA JET
2099 DATA SCRIVERE
2100 DATA LEGGERE
2101 DATA CALIFORNIA
2102 DATA AMICO
2103 DATA SAPORI
2104 DATA PROGRAMMA
2105 DATA DEBUG
2106 DATA DISK
2107 DATA MUSICA
2108 DATA CANTANTE
2109 DATA TELEFONO
2110 DATA TELEGRAFO
2111 DATA TELEX
2112 DATA CASSETTA
2113 DATA LOCALI
2114 DATA NETWORK
2115 DATA JAZZ
2116 DATA STATI
2117 DATA FORMULA
2118 DATA BORSA
2119 DATA PARTICELLA
2120 DATA PROTONE
2121 DATA QUARK
2122 DATA NEUTRINO
2123 DATA ELETTRONE
2124 DATA ACQUA
2125 DATA ARANCIA
2126 DATA ANANAS
2127 DATA POMODORO
2128 DATA KIWI
2129 DATA TESTA
2130 DATA BRACCIO
2131 DATA GAMBA
2132 DATA OCCHI
2133 DATA NASO
2134 DATA CUORE
2135 DATA UDIRE
2136 DATA VEDERE
2137 DATA GIACCA
2138 DATA CRAVATTA
2139 DATA METRO
2140 DATA INCH
2141 DATA CACCIAVITE
2142 DATA CARBURATORE
2143 DATA STETOSCOPIO
2144 DATA PAROLE
2145 DATA PAGINE
2146 DATA CONTENITORE
2147 DATA DECOLLO
2148 DATA DESIGN
2149 DATA ARTISTA
2150 DATA RAGIONE
2151 DATA NASTRO
2152 DATA MICROFONO
2153 DATA STRUMENTO
2154 DATA PIANOFORTE
2155 DATA CLARINETTO
2156 DATA SAX
2157 DATA SYNTHESIZER
2158 DATA ECONOMIA
2159 DATA MONETA
2160 DATA DOLLARO
2161 DATA STERLINA
2162 DATA CAMBIO

2163 DATA SOFFITTA
2164 DATA PANINO
2165 DATA SPAGHETTI
2166 DATA HAMBURGER
2167 DATA WHISKY
2168 DATA CANCELLARE
2169 DATA MEMORIA
2170 DATA AUTOBUS
2171 DATA COMBINAZIONE
2172 DATA URAGANO
2173 DATA SPACE SHUTTLE
2174 DATA ASTRONAUTA
2175 DATA PIOGGIA
2176 DATA CAMINETTO
2177 DATA STRADA
2178 DATA PERCORRERE
2179 DATA INDUSTRIA
2180 DATA INGRANAGGIO
2181 DATA RIVISTA
2182 DATA UNIX
2183 DATA BLOCCHETTO
2184 DATA DINOSAURO
2185 DATA DIZIONARIO
2186 DATA VIDEOGAMES
2187 DATA STAR WARS
2188 DATA PRODUZIONE
2189 DATA ALLENAMENTO
2190 DATA BASKET
2191 DATA DISPLAY
2192 DATA PIXEL
2193 DATA LOGARITMO
2194 DATA TRIGONOMETRIA
2195 DATA HOBBY
2196 DATA TELEFILM
2197 DATA NOTIZIE
2198 DATA SOPRANO
2199 DATA PENULTIMA
2200 DATA ULTIMA
9998 DATA #
9999 REM FINE LISTA

CRONOMETRO

Il **Cronometro** per **Atari 130** permette di misurare i tempi e di visualizzarli in ore, minuti, secondi e decimi di secondo. Il cronometro è controllato dai tasti Z (Start), X (Lap, tempo intermedio) e C (Stop).

```
1000 REM * * *      CRONOMETRO      * * *
1010 REM -----
1020 REM           di Gaetano Marano   -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 REM           funzione           tasto   -
1070 REM -----
1080 REM           START              Z       -
1090 REM           LAP                X       -
1100 REM           STOP               C       -
1110 REM -----
1120 GRAPHICS 0
1130 SOUND 0.0,0.0: SOUND 1.0,0.0
1140 SOUND 2.0,0.0: SOUND 3.0,0.0
1150 POKE 710,0: POKE 712,2: POKE 709,10
1160 POKE 82,0: POKE 752,1
1170 REM -----
1175 DIM H$(2), M$(2), S$(2), C$(2)
1180 POSITION 0,2: PRINT "  ||||| "
1190 POSITION 0,3: PRINT "  ||||| "
1200 POSITION 0,4: PRINT "  ||||| "
1210 POSITION 13,1: PRINT "hr min sec"
1220 POSITION 5,8: PRINT "tasti: Z=START X=LAP C=STOP"
1230 POSITION 13,3: PRINT "00: 00: 00. 0"
1240 K=PEEK(764)
1245 POKE 764,0
1250 IF K=23 THEN 1290
1270 IF K=22 THEN 1310
1280 GOTO 1240
1290 POKE 18,0: POKE 19,0: POKE 20,0
1300 POSITION 13,3: PRINT "00: 00: 00. 0"
1310 GOSUB 1370
1320 K=PEEK(764)
1330 POKE 764,0
1340 IF K=18 THEN POKE 18,0: POKE 19,0: POKE 20,0: GOTO 1240
1350 IF K=22 THEN K=5: GOTO 1240
1360 GOTO 1310
1370 T=INT(PEEK(18)*65535+PEEK(19)*256+PEEK(20))/5
1380 H=INT(T/3600)
1390 T=T-H*3600
1400 M=INT(T/60)
1410 T=T-M*60
1420 S=INT(T/10)
1430 C=INT((T-S*10))
1440 H$=STR$(H)
1450 M$=STR$(M)
```

```
1460 S#=STR$(S)
1470 C#=STR$(C)
1480 POSITION 15-LEN(H#),3:PRINT H#
1490 POSITION 19-LEN(M#),3:PRINT M#
1500 POSITION 23-LEN(S#),3:PRINT S#
1510 POSITION 25,3:PRINT C#
1515 IF M<10 THEN POSITION 17,3:PRINT "0"
1516 IF S<10 THEN POSITION 21,3:PRINT "0"
1520 RETURN
```

SCRITTE PUBBLICITARIE

Il programma chiede di inserire per prima cosa un testo di massimo 100 caratteri che può contenere, ad esempio, un messaggio pubblicitario, una serie di prezzi, delle quotazioni di borsa, delle offerte speciali, dei comunicati importanti, e così via.

Quindi si deve inserire la velocità di rotazione delle scritte con un valore tra 1 (più veloce) a 10 (più lenta).

Per inserire una frase diversa occorre fermare il programma con il tasto BREAK e quindi dare di nuovo RUN.

```
1000 REM * * SCRITTE PUBBLICITARIE * *
1010 REM -----
1020 REM - di Gaetano Marano -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1080 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 710,0:POKE 712,0:POKE 709,10
1095 POKE 82,0:POKE 752,0
1096 REM -----
1100 DIM X$(1):X$=CHR$(146)
1110 DIM A$(120),S$(40),C$(160),H$(47)
1120 S$="
1125 POSITION 0,0
1130 PRINT "Inserire frase (max 100 caratteri) e premere RETURN"
1135 INPUT A$
1136 POKE 752,1
1140 PRINT ""
1150 PRINT "Inserire velocità (1--10) e premere RETURN"
1155 INPUT T
1158 IF T<1 OR T>10 THEN 1150
1160 PRINT ""
1165 FOR A=1 TO 38
1170 POSITION A,4:PRINT X$
1180 POSITION A,5:PRINT X$
1190 POSITION A,9:PRINT X$
1200 POSITION A,10:PRINT X$
1205 NEXT A
1220 IF LEN(A$)<38 THEN H$=S$(1,48-LEN(A$))
1230 IF LEN(A$)>37 THEN H$=S$(1,10)
1235 Q=LEN(A$):H=LEN(H$)
1240 A$(Q+1,Q+H)=H$
1250 C$=A$
1260 POSITION 1,7:PRINT C$(1,38)
1270 C$=C$(2,LEN(C$))
1280 IF LEN(C$)<39 THEN C$=C$(39,Q+H)=A$
1290 FOR D=1 TO 20* T:NEXT D
1300 GOTO 1260
```


CONTAPEZZI

Un uso insolito del computer può essere quello di Contapezzi capace di contare da 10 a 99.999.999 pezzi.

Facendo girare il programma, ogni volta che si preme RETURN il numero viene incrementato di uno. Oltre a ciò, si può inserire anche un numero qualsiasi da aggiungere (per esempio 334925) che premendo RETURN viene sommato al numero presente sullo schermo oppure sottratto, se il numero inserito è preceduto dal segno.

Se si cerca di sommare o di sottrarre un numero che farebbe scendere il totale sotto lo 0 o lo farebbe salire oltre il massimo, il programma fa uscire sul video la frase **"NUMERO ERRATO O ECCESSIVO"**.

```
1000 REM * * *      CONTAPEZZI      * * *
1010 REM -----
1020 REM -      di Gaetano Marano  -
1030 REM -----
1040 REM <c> Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 709,10:POKE 710,0:POKE 712,0
1093 POKE 82,0:POKE 752,1
1095 REM -----
1100 DIM A$(40),N$(10),P$(10),W$(8)
1101 DIM B$(1),C$(1),D$(1),E$(1),F$(1),G$(1),H$(1),I$(1),L$(1),M$(1),O$(1),Q$(1)
1103 B$=CHR$(17):C$=CHR$(22):D$=CHR$(26):E$=CHR$(5):F$=CHR$(2):G$=CHR$(3)
1104 H$=CHR$(12):I$=CHR$(1):L$=CHR$(4):M$=CHR$(124):O$=CHR$(23):Q$=CHR$(24)
1110 POSITION 0,0
1120 PRINT " Premere RETURN per un solo pezzo"
1130 PRINT
1140 PRINT " o introdurre N";H$;" pezzi in piu' o"
1150 PRINT
1160 PRINT " in meno (con -) e premere RETURN."
1170 PRINT
1190 POSITION 2,12:PRINT B$;E$;" ";B$;E$;" ";B$;E$;" ";B$;E$;" ";O$
1200 POSITION 2,13:PRINT I$;G$;" ";I$;" ";B$;G$;" ";B$;G$;" ";M$
1210 POSITION 2,14:PRINT M$;" ";D$;G$;" ";D$;G$;" ";D$;G$;" ";Q$
1220 N=0
1230 C=0
1240 POSITION 0,22:INPUT N$
```

```

1250 POSITION 0,22:PRINT "
1260 IF N$="" THEN N$="1"
1270 IF ASC(N$)<>45 THEN P$=N$:FOR A=1 TO LEN(N$)
1280 IF ASC(N$)=45 THEN P$=N$(2,LEN(N$)):FOR A=2 TO LEN(N$)
1290 IF ASC(P$)<48 OR ASC(P$)>57 THEN 1440
1295 IF LEN(P$)>1 THEN P$=P$(2,LEN(P$))
1300 NEXT A
1310 IF (N+VAL(N$))<(10000000-1) OR (N+VAL(N$))<0 THEN 1440
1320 N=N+VAL(N$)
1327 N$="1"
1330 A$=STR$(N)
1340 L=LEN(A$)
1345 FOR P=1 TO 8
1350 W$(P)="!"
1352 IF P<=L THEN W$(P)=A$(P)
1355 NEXT P
1356 A$=W$
1360 FOR P=1 TO 8
1370 C=P*2+17
1380 J=ASC(A$)-47
1390 IF J=-14 THEN J=11
1400 ON J GOSUB 1480,1520,1560,1600,1640,1680,1720,1760,1800,1840,1880
1410 IF LEN(A$)>1 THEN A$=A$(2,LEN(A$))
1420 NEXT P
1430 GOTO 1230
1440 POSITION 6,18:PRINT " NUMERO ERRATO O ECCESSIVO "
1450 FOR T=1 TO 500:NEXT T
1460 POSITION 6,18:PRINT "
1470 GOTO 1230
1480 POSITION C,12:PRINT B$;E$
1490 POSITION C,13:PRINT M$;M$
1500 POSITION C,14:PRINT D$;G$
1510 RETURN
1520 POSITION C,12:PRINT " ";E$
1530 POSITION C,13:PRINT " ";M$
1540 POSITION C,14:PRINT " ";Q$
1550 RETURN
1560 POSITION C,12:PRINT B$;E$
1570 POSITION C,13:PRINT B$;G$
1580 POSITION C,14:PRINT D$;G$
1590 RETURN
1600 POSITION C,12:PRINT B$;E$
1610 POSITION C,13:PRINT " ";L$
1620 POSITION C,14:PRINT D$;G$
1630 RETURN
1640 POSITION C,12:PRINT M$;M$
1650 POSITION C,13:PRINT D$;L$
1660 POSITION C,14:PRINT " ";M$
1670 RETURN
1680 POSITION C,12:PRINT B$;E$
1690 POSITION C,13:PRINT D$;E$
1700 POSITION C,14:PRINT D$;G$
1710 RETURN
1720 POSITION C,12:PRINT B$;E$
1730 POSITION C,13:PRINT I$;E$
1740 POSITION C,14:PRINT D$;G$
1750 RETURN
1760 POSITION C,12:PRINT B$;E$
1770 POSITION C,13:PRINT " ";M$
1780 POSITION C,14:PRINT " ";M$
1790 RETURN
1800 POSITION C,12:PRINT B$;E$
1810 POSITION C,13:PRINT I$;L$
1820 POSITION C,14:PRINT D$;G$
1830 RETURN
1840 POSITION C,12:PRINT B$;E$
1850 POSITION C,13:PRINT D$;L$
1860 POSITION C,14:PRINT D$;G$
1870 RETURN
1880 POSITION C,12:PRINT " "
1890 POSITION C,13:PRINT " "
1900 POSITION C,14:PRINT " "
1910 RETURN

```


FISICA DEI REATTORI NUCLEARI

Il programma simula il processo di “fissione nucleare” che è alla base del funzionamento appunto delle centrali nucleari che producono energia.

La fissione nucleare è una reazione provocata da neutroni veloci che colpiscono degli atomi di Uranio 235 i quali si spaccano producendo due atomi più piccoli di Bario e Krypton ed altri due neutroni che colpendo a loro volta degli atomi di Uranio provocano la cosiddetta “reazione a catena”. La reazione produce anche energia che viene sfruttata per produrre calore ed elettricità.

Il programma produce un’animazione che può essere vista in modo lento (premendo il tasto L) o in modo veloce (tasto V), il tasto R fa ritornare il programma allo stato iniziale.

```
1000 REM FISICA DEI REATTORI NUCLEARI
1010 REM -----
1020 REM - di Gaetano Marano -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0: SOUND 1,0,0,0
1080 SOUND 2,0,0,0: SOUND 3,0,0,0
1090 POKE 709,10: POKE 710,0: POKE 712,114
1100 POKE 82,0: POKE 752,1
1110 REM -----
1120 DIM X$(1): X$=CHR$(20)
1130 DIM Z$(1): Z$=CHR$(18)
1140 POSITION 6,0: PRINT "SIMULATORE FISSIONE NUCLEARE"
1150 FOR A=0 TO 39
1155 POSITION A,1: PRINT Z$
1158 NEXT A
1160 POSITION 1,2: PRINT "L_Lento"
1170 POSITION 1,4: PRINT "V_Veloce"
1180 POSITION 1,6: PRINT "R_Ripeti"
1190 POSITION 27,2: PRINT "._Neutrone"
1200 POSITION 27,3: PRINT "0_Uranio 235"
```

```

1210 POSITION 27,4:PRINT "o_Bario 139"
1220 POSITION 27,5:PRINT X$;"_Kryeto 97"
1230 POSITION 19,2:PRINT "."
1240 POSITION 19,6:PRINT "0"
1250 POSITION 12,13:PRINT "0"
1260 POSITION 26,13:PRINT "0"
1270 POSITION 7,18:PRINT "0"
1280 POSITION 17,18:PRINT "0"
1290 POSITION 31,18:PRINT "0"
1300 POSITION 31,18:PRINT "0"
1305 W=PEEK(764)
1310 POKE 764,255
1330 IF W=0 THEN T=60:GOTO 1350
1340 IF W=16 THEN T=1:GOTO 1350
1345 GOTO 1305
1350 FOR A=3 TO 6
1360 GOSUB 1720
1370 POSITION 19,A:PRINT "."
1380 POSITION 19,A-1:PRINT " "
1390 NEXT A
1400 POSITION 18,6:PRINT "o ";X$
1410 FOR A=7 TO 13
1420 GOSUB 1720
1430 POSITION 25-A,A:PRINT "."
1440 POSITION 13+A,A:PRINT "."
1450 POSITION 26-A,A-1:PRINT " "
1460 POSITION 12+A,A-1:PRINT " "
1470 NEXT A
1480 POSITION 11,13:PRINT "o ";X$
1490 POSITION 25,13:PRINT "o ";X$
1500 FOR A=14 TO 18
1510 GOSUB 1720
1520 POSITION 39-A,A:PRINT "."
1530 POSITION 25-A,A:PRINT "."
1540 POSITION 13+A,A:PRINT "."
1550 POSITION -1+A,A:PRINT "."
1560 POSITION 40-A,A-1:PRINT " "
1570 POSITION 26-A,A-1:PRINT " "
1580 POSITION 12+A,A-1:PRINT " "
1590 POSITION -2+A,A-1:PRINT " "
1600 NEXT A
1610 POSITION 6,18:PRINT "o ";X$;"          o ";X$;" o ";X$;"          o ";X$"
1620 GOSUB 1720
1630 POSITION 6,19:PRINT ". . . . . ."
1640 GOSUB 1720
1650 POSITION 5,19:PRINT " "
1660 POSITION 5,20:PRINT ". . . . ."
1670 GOSUB 1720
1680 POSITION 5,20:PRINT " "
1685 POSITION 4,21:PRINT ". . . . ."
1690 GOSUB 1720
1695 POSITION 4,21:PRINT " "
1700 W=PEEK(764)
1705 POKE 764,255
1700 IF W=40 THEN RUN
1710 GOTO 1690
1720 FOR D=1 TO T
1730 NEXT D:RETURN

```

PIANO 3—OTTAVE

Il programma trasforma un computer 130XE in un semplice piano a 3 ottave per un totale di 37 note che possono essere suonate tramite 25 tasti più il tasto di SHIFT.

I tasti che producono le note sono indicati nella tabella che segue (il segno + indica il diesis).

Tasto	Nota	Ottava
Q	DO	1
2	DO+	1
W	RE	1
3	RE+	1
E	MI	1
R	FA	1
5	FA+	1
T	SOL	1
6	SOL+	1
Y	LA	1
7	LA+	1
U	SI	1
Z	DO	2
S	DO+	2
X	RE	2
D	RE+	2
C	MI	2
V	FA	2
G	FA+	2
B	SOL	2
H	SOL+	2
N	LA	2

/segue/

J	LA+	2
M	SI	2
,	DO	3
SHIFT+Z	DO	3
SHIFT+S	DO+	3
SHIFT+X	RE	3
SHIFT+D	RE+	3
SHIFT+C	MI	3
SHIFT+V	FA	3
SHIFT+G	FA+	3
SHIFT+B	SOL	3
SHIFT+H	SOL+	3
SHIFT+N	LA	3
SHIFT+J	LA+	3
SHIFT+M	SI	3
SHIFT+,	DO	4

```

1000 REM **      PIANO 3_OTTAVE      **
1010 REM -----
1020 REM -      Di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1080 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 710,112:POKE 712,22
1100 POKE 82,0:POKE 752,1:POKE 709,10
1110 REM -----
1120 DIM N$(5)
1130 POSITION 4,1:PRINT " PIANO 3_OTTAVE "
1140 POSITION 8,8:PRINT "OTTAVA -- "
1150 POSITION 8,10:PRINT " NOTA -- "
1200 POKE 764,0
1240 K=PEEK(764)
1250 POKE 764,0
1260 IF K<127 THEN GOSUB K+2000
1270 SOUND 1,N,10,15
1280 POSITION 18,8:PRINT N$(1,1)
1290 POSITION 18,10:PRINT N$(2,5)
1295 FOR T=1 TO 10:NEXT T
1300 GOTO 1240
2000 N$="":N=0:RETURN
2001 N$="2LA#":N=68:RETURN
2002 N$="":N=0:RETURN
2003 N$="":N=0:RETURN
2004 N$="":N=0:RETURN
2005 N$="":N=0:RETURN
2006 N$="":N=0:RETURN
2007 N$="":N=0:RETURN
2008 N$="":N=0:RETURN
2009 N$="":N=0:RETURN

```

```

2010 N$=" " " N=0: RETURN
2011 N$="1SI " N=128: RETURN
2012 N$=" " " N=0: RETURN
2013 N$=" " " N=0: RETURN
2014 N$=" " " N=0: RETURN
2015 N$=" " " N=0: RETURN
2016 N$="2FA " N=91: RETURN
2017 N$=" " " N=0: RETURN
2018 N$="2MI " N=96: RETURN
2019 N$=" " " N=0: RETURN
2020 N$=" " " N=0: RETURN
2021 N$="2SOL " N=81: RETURN
2022 N$="2RE " N=108: RETURN
2023 N$="2D0 " N=121: RETURN
2024 N$=" " " N=0: RETURN
2025 N$=" " " N=0: RETURN
2026 N$="1RE# " N=204: RETURN
2027 N$="1SOL# " N=153: RETURN
2028 N$=" " " N=0: RETURN
2029 N$="1FA# " N=173: RETURN
2030 N$="100# " N=230: RETURN
2031 N$=" " " N=0: RETURN
2032 N$="3D0 " N=60: RETURN
2033 N$=" " " N=0: RETURN
2034 N$=" " " N=0: RETURN
2035 N$="2LA " N=72: RETURN
2036 N$=" " " N=0: RETURN
2037 N$="2SI " N=64: RETURN
2038 N$=" " " N=0: RETURN
2039 N$=" " " N=0: RETURN
2040 N$="1FA " N=182: RETURN
2041 N$=" " " N=0: RETURN
2042 N$="1MI " N=193: RETURN
2043 N$="1LA " N=144: RETURN
2044 N$=" " " N=0: RETURN
2045 N$="1SOL " N=162: RETURN
2046 N$="1RE " N=217: RETURN
2047 N$="100 " N=243: RETURN
2048 N$=" " " N=0: RETURN
2049 N$=" " " N=0: RETURN
2050 N$=" " " N=0: RETURN
2051 N$="1LA# " N=136: RETURN
2052 N$=" " " N=0: RETURN
2053 N$=" " " N=0: RETURN
2054 N$=" " " N=0: RETURN
2055 N$=" " " N=0: RETURN
2056 N$=" " " N=0: RETURN
2057 N$="2SOL# " N=76: RETURN
2058 N$="2RE# " N=102: RETURN
2059 N$=" " " N=0: RETURN
2060 N$=" " " N=0: RETURN
2061 N$="2FA# " N=85: RETURN
2062 N$="2D0# " N=114: RETURN
2063 N$=" " " N=0: RETURN
2064 N$=" " " N=0: RETURN
2065 N$="3LA# " N=33: RETURN
2066 N$=" " " N=0: RETURN
2067 N$=" " " N=0: RETURN
2068 N$=" " " N=0: RETURN
2069 N$=" " " N=0: RETURN
2070 N$=" " " N=0: RETURN
2071 N$=" " " N=0: RETURN
2072 N$=" " " N=0: RETURN
2073 N$=" " " N=0: RETURN
2074 N$=" " " N=0: RETURN
2075 N$=" " " N=0: RETURN
2076 N$=" " " N=0: RETURN
2077 N$=" " " N=0: RETURN
2078 N$=" " " N=0: RETURN
2079 N$=" " " N=0: RETURN
2080 N$="3FA " N=45: RETURN
2081 N$=" " " N=0: RETURN

```

```

2082 N$="3MI" ":N=47:RETURN
2083 N$=" ":N=0:RETURN
2084 N$=" ":N=0:RETURN
2085 N$="3SOL":N=40:RETURN
2086 N$="3RE":N=53:RETURN
2087 N$="3DO":N=60:RETURN
2088 N$=" ":N=0:RETURN
2089 N$=" ":N=0:RETURN
2090 N$=" ":N=0:RETURN
2091 N$=" ":N=0:RETURN
2092 N$=" ":N=0:RETURN
2093 N$=" ":N=0:RETURN
2094 N$=" ":N=0:RETURN
2095 N$=" ":N=0:RETURN
2096 N$="400":N=29:RETURN
2097 N$=" ":N=0:RETURN
2098 N$=" ":N=0:RETURN
2099 N$="3LA":N=35:RETURN
2100 N$=" ":N=0:RETURN
2101 N$="3SI":N=31:RETURN
2102 N$=" ":N=0:RETURN
2103 N$=" ":N=0:RETURN
2104 N$=" ":N=0:RETURN
2105 N$=" ":N=0:RETURN
2106 N$=" ":N=0:RETURN
2107 N$=" ":N=0:RETURN
2108 N$=" ":N=0:RETURN
2109 N$=" ":N=0:RETURN
2110 N$=" ":N=0:RETURN
2111 N$=" ":N=0:RETURN
2112 N$=" ":N=0:RETURN
2113 N$=" ":N=0:RETURN
2114 N$=" ":N=0:RETURN
2115 N$=" ":N=0:RETURN
2116 N$=" ":N=0:RETURN
2117 N$=" ":N=0:RETURN
2118 N$=" ":N=0:RETURN
2119 N$=" ":N=0:RETURN
2120 N$=" ":N=0:RETURN
2121 N$="3SOL#":N=37:RETURN
2122 N$="3RE#":N=50:RETURN
2123 N$=" ":N=0:RETURN
2124 N$=" ":N=0:RETURN
2125 N$="3FA#":N=42:RETURN
2126 N$="3DO#":N=57:RETURN

```

PROGRAMMATORE DI SUONI

Una caratteristica interessante del 130XE è certamente il generatore di suoni a 4 voci.

Con il Programmatore di Suoni è possibile modificare facilmente e velocemente i parametri delle istruzioni che determinano le caratteristiche del suono fino ad arrivare al suono desiderato.

Usando i tasti di cursore a destra e a sinistra è possibile selezionare la voce da modificare, mentre tramite i tasti cursore in alto e in basso è possibile scegliere il parametro da modificare (i tasti di cursore vanno premuti insieme al tasto Control).

È possibile scegliere anche il valore dell'incremento o del decremento tramite i tasti U (uno), D (dieci) e C (cento); è importante tenere presente che se si cerca di decrementare un registro che si trova a 15 fino al valore di 8 dopo aver selezionato un decremento, ad esempio di cento, il decremento non avviene finché non si preme il tasto U per ritornare al decremento uno. Inoltre gli incrementi o decrementi del parametro Distorsione avvengono di 2 in 2 (da 0 a 14 e viceversa) in quanto i valori dispari non hanno effetto.

Il parametro selezionato può essere incrementato o decrementato tramite i tasti + e —; oltre a questi c'è il tasto A che consente di ascoltare il suono creato fino a quel momento e di avere anche le relative istruzioni SOUND ed il tasto ESC che azzerà tutto e fa ripartire il programma.

Da notare che all'inizio può sembrare che il programma non funzioni, infatti, per sentire ad esempio un suono dalla voce

1, occorre prima alzare il volume del televisore o del monitor, quindi stabilire un volume sufficiente, inserire una frequenza per il suono e premere il tasto A.

Una descrizione più dettagliata delle istruzioni suono è presente nel manuale del computer **Atari 130XE**.

```

1000 REM *   PROGRAMMATORE DI SUONI *
1010 REM -----
1020 REM -       Di Gaetano Marano       -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 710,112:POKE 709,10
1100 POKE 82,0:POKE 752,1:POKE 712,112
1110 REM -----
1120 DIM A$(17),B$(17),C$(17)
1130 FOR L=1 TO 17
1140 A$(L)=CHR$(18)
1145 IF L=5 OR L=9 OR L=13 THEN A$(L)=CHR$(23)
1150 B$(L)=CHR$(18)
1155 IF L=5 OR L=9 OR L=13 THEN B$(L)=CHR$(19)
1160 C$(L)=CHR$(18)
1165 IF L=5 OR L=9 OR L=13 THEN C$(L)=CHR$(24)
1170 NEXT L
1180 A$(1)=CHR$(17):A$(17)=CHR$(5)
1190 B$(1)=CHR$(1):B$(17)=CHR$(4)
1200 C$(1)=CHR$(26):C$(17)=CHR$(3)
1210 POSITION 2,0:PRINT " PROGRAMMATORE DI SUONI "
1220 POSITION 4,5:PRINT "VOCE ----- 0 1 2 3 "
1230 POSITION 18,6:PRINT A$
1240 POSITION 4,7:PRINT "FREQUENZA --- 10 10 10 10 1"
1250 POSITION 18,8:PRINT B$
1260 POSITION 4,9:PRINT "DISTRORSIONE _ 10 10 10 10 1"
1270 POSITION 18,10:PRINT B$
1280 POSITION 4,11:PRINT "VOLUME ----- 10 10 10 10 1"
1290 POSITION 18,12:PRINT C$
1300 DIM D(11,3)
1302 FOR A=7 TO 11
1303 FOR B=0 TO 3
1304 D(A,B)=0
1305 NEXT B
1306 NEXT A
1310 POSITION 35,7:PRINT "<"
1330 Y=7:I=1:V=0
1350 K=PEEK(764)
1360 POKE 764,0
1380 IF K=142 OR K=143 THEN GOSUB 1500
1400 IF K=28 THEN RUN
1410 IF K=11 THEN GOSUB 2150
1420 IF K=58 THEN GOSUB 2160
1430 IF K=18 THEN GOSUB 2170
1440 IF K=63 THEN GOSUB 1900
1450 IF K=134 OR K=135 THEN GOSUB 2200
1460 IF K=6 OR K=14 THEN GOSUB 1560
1490 GOTO 1350
1500 H=Y
1510 IF K=142 THEN Y=Y-2
1520 IF K=143 THEN Y=Y+2
1530 IF Y=13 THEN Y=7
1540 IF Y=5 THEN Y=11
1545 POSITION 35,H:PRINT " "
1546 POSITION 35,Y:PRINT "<"
1550 RETURN
1560 IF Y=7 AND K=14 AND D(Y,V)-I=0 THEN D(Y,V)=D(Y,V)-I

```



```

1570 IF Y=7 AND K=6 AND D(Y,V)+I<256 THEN D(Y,V)=D(Y,V)+I
1580 IF Y=9 AND K=14 AND D(Y,V)-I*2=0 THEN D(Y,V)=D(Y,V)-I*2
1590 IF Y=9 AND K=6 AND D(Y,V)+I*2<15 THEN D(Y,V)=D(Y,V)+I*2
1600 IF Y=11 AND K=14 AND D(Y,V)-I=0 THEN D(Y,V)=D(Y,V)-I
1610 IF Y=11 AND K=6 AND D(Y,V)+I<16 THEN D(Y,V)=D(Y,V)+I
1620 GOSUB 1850
1840 RETURN
1850 POSITION V*4+19,Y:PRINT " "
1860 POSITION V*4+19,Y:PRINT D(Y,V)
1870 RETURN
1900 SOUND 0,D(7,0),D(9,0),D(11,0)
1910 SOUND 1,D(7,1),D(9,1),D(11,1)
1920 SOUND 2,D(7,2),D(9,2),D(11,2)
1930 SOUND 3,D(7,3),D(9,3),D(11,3)
1935 POSITION 4,14:PRINT "SOUND 0,"
1940 POSITION 4,14:PRINT "SOUND 0,";D(7,0);", ";D(9,0);", ";D(11,0)
1945 POSITION 4,16:PRINT "SOUND 1,"
1950 POSITION 4,16:PRINT "SOUND 1,";D(7,1);", ";D(9,1);", ";D(11,1)
1955 POSITION 4,18:PRINT "SOUND 2,"
1960 POSITION 4,18:PRINT "SOUND 2,";D(7,2);", ";D(9,2);", ";D(11,2)
1965 POSITION 4,20:PRINT "SOUND 3,"
1970 POSITION 4,20:PRINT "SOUND 3,";D(7,3);", ";D(9,3);", ";D(11,3)
1980 RETURN
2150 I=1:RETURN
2160 I=10:RETURN
2170 I=100:RETURN
2200 IF K=134 THEN V=V-1
2210 IF K=135 THEN V=V+1
2220 IF V=-1 THEN V=3
2230 IF V=4 THEN V=0
2240 IF V=0 THEN POSITION 4,5:PRINT "VOCE ----- 0 1 2 3 "
2250 IF V=1 THEN POSITION 4,5:PRINT "VOCE ----- 0 1 2 3 "
2260 IF V=2 THEN POSITION 4,5:PRINT "VOCE ----- 0 1 2 3 "
2270 IF V=3 THEN POSITION 4,5:PRINT "VOCE ----- 0 1 2 3 "
2280 RETURN

```


SIRENA AMERICANA

Usando il generatore di suoni del 130XE, il programma produce il suono di una sirena Americana.

```
1000 REM * *      SIRENA AMERICANA      * *
1010 REM -----
1020 REM -      di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 710,0:POKE 712,0:POKE 709,10
1100 POKE 82,0:POKE 752,1
1110 REM -----
1120 PRINT
1130 PRINT "      SIRENA AMERICANA"
1170 FOR A=9 TO 30
1180 SOUND 1,A,14,15
1190 NEXT A
1200 FOR A=30 TO 9 STEP -1
1210 SOUND 1,A,14,15
1220 NEXT A
1230 GOTO 1170
```


CARATTERI IN GRAPHICS 24

Questo è sicuramente una delle routine più utili per chi vuole realizzare programmi in BASIC con grafica e testo su un computer **Atari 130XE**.

Infatti questo computer dispone di numerosi modi solo grafici, modi solo testo e modi di grafica con finestra testo (le quattro righe più in basso) ma non consente di inserire dei testi direttamente nell'area grafica.

Poter aggiungere un testo è particolarmente utile nel modo grafico a più alta risoluzione (329 x 192 pixel) e senza finestra testo (**GRAPHICS 24**) come dimostrato anche da alcuni programmi di questo libro.

La routine per inserire dei testi in GRAPHICS 24 (linee da 8000 a 8190) è molto semplice, infatti le forme relative al set di caratteri del **130XE** sono memorizzate sotto forma di codici nella ROM, a partire dalla locazione 57344, e possono essere letti con PEEK e scritti direttamente nella memoria video con l'istruzione POKE.

Per scrivere uno o più testi nel modo grafico occorre OGNI VOLTA inserire il testo nella stringa A\$, stabilire la colonna e la linea tramite le variabili CX ed LX (usando numeri interi e non decimali) e quindi eseguire un'istruzione GOSUB 8000. Le istruzioni da 100 a 360 sono un programma dimostrativo del funzionamento della routine e possono essere tolte (tranne la linea 120) se si usa la routine in un'altro programma.

```

100 REM * Programma dimostrativo *
110 REM -----
120 GRAPHICS 24
130 SOUND 0,0,0,0:SOUND 1,0,0,0
140 SOUND 2,0,0,0:SOUND 3,0,0,0
150 POKE 712,116:POKE 710,0:POKE 709,10
160 POKE 82,0
180 REM -----
190 REM a$ stringa caratteri
200 REM cx 0...39 colonna
210 REM lx 0...23 linea
240 REM -----
250 DIM A$(40),T$(32)
255 FOR D=0 TO 7
260 T$=""
270 FOR T=1 TO 32
280 T$(T)=CHR$(D*32+T-1):NEXT T
300 A$=T$
310 CX=3: LX=D*2+4
330 GOSUB 8000
340 NEXT D
350 GOTO 350
360 REM -----
8000 REM * CARATTERI IN GRAPHICS 24 *
8010 REM -----
8020 REM - di Gaetano Marano -
8030 REM -----
8040 REM (c) Gruppo Editoriale Jackson
8050 REM -----
8060 IF A$="" OR LX<0 OR LX>23 OR CX<0 OR CX>39 THEN RETURN
8070 SA=PEEK(560)+PEEK(561)*256+282
8080 FOR K=1 TO LEN(A$)
8090 J=ASC(A$(K))
8095 M=0: IF J>127 THEN J=J-128: M=1
8096 IF J>95 THEN GOTO 8120
8100 IF J<32 THEN J=J+64: GOTO 8120
8110 IF J>31 AND J<96 THEN J=J-32
8120 CA=J*8
8130 ST=LX*320+CX+SA
8140 FOR P=0 TO 7
8150 IF M=0 THEN POKE ST+P*40,PEEK(57344+P+CA)
8155 IF M=1 THEN POKE ST+P*40,255-PEEK(57344+P+CA)
8160 NEXT P
8170 CX=CX+1
8180 NEXT K
8190 RETURN

```

TIRO ALLA FUNE

Il **tiro alla fune** è un gioco per due persone.

All'inizio appaiono sullo schermo due omini stilizzati con una fune.

L'omino di sinistra viene comandato dal tasto A mentre quello di destra dal tasto L. Ogni volta che appaiono le due stelle al centro dello schermo il giocatore che preme per primo il proprio tasto ha la possibilità di tirare i due omini dalla sua parte, se però uno dei giocatori preme il proprio tasto prima che appaiono le stelle, i due omini si sposteranno nel senso opposto favorendo l'avversario.

Il gioco continua finché uno dei due giocatori non riesce a trascinare l'avversario completamente nella propria zona.

Il tempo di apparizione delle due stelle ed il ritardo tra una apparizione e l'altra sono casuali e vengono stabiliti rispettivamente dalle linee 1290 e 1230 che possono essere eventualmente modificate per rendere il gioco più difficile.

```
1000 REM * * * TIRO ALLA FUNE * * *
1010 REM -----
1020 REM - di Gaetano Marano -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 709,10:POKE 710,0:POKE 712,114
1100 POKE 82,0:POKE 752,1
1110 REM -----
1111 DIM A$(20),B$(20),C$(20),D$(20),R$(1),W$(40)
1114 FOR A=1 TO 40
1115 W$(A)=CHR$(14)
1116 NEXT A
1117 FOR A=3 TO 18
1118 B$(A)=CHR$(18):D$(A)=CHR$(14)
1119 A$(A)=CHR$(32):C$(A)=CHR$(32)
1120 NEXT A
1121 A$(1)=CHR$(20):A$(2)=CHR$(32):A$(19)=CHR$(32):A$(20)=CHR$(111)
```

```

1122 B$(1)=CHR$(19):B$(2)=CHR$(18):B$(19)=CHR$(18):B$(20)=CHR$(19)
1123 C$(1)=CHR$(1):C$(2)=CHR$(5):C$(19)=CHR$(17):C$(20)=CHR$(4)
1124 D$(1)=CHR$(26):D$(2)=CHR$(26):D$(19)=CHR$(3):D$(20)=CHR$(3)
1125 POSITION 0,9:PRINT W$
1126 POSITION 10,6:PRINT A$
1130 POSITION 10,7:PRINT B$
1140 POSITION 10,8:PRINT C$
1150 POSITION 10,9:PRINT D$
1170 POSITION 1,1:PRINT CHR$(17);CHR$(18);CHR$(5)
1175 POSITION 36,1:PRINT CHR$(17);CHR$(18);CHR$(5)
1180 POSITION 1,2:PRINT "IAI"
1190 POSITION 1,3:PRINT CHR$(26);CHR$(18);CHR$(3)
1195 POSITION 36,3:PRINT CHR$(26);CHR$(18);CHR$(3)
1200 P=9
1210 M=-1
1220 POSITION 19,4:PRINT " "
1230 FOR A=1 TO 40+RND(0)*100
1232 W=PEEK(764)
1234 POKE 764,255
1240 IF W=63 OR W=0 THEN GOTO 1315
1260 NEXT A
1270 M=1
1280 POSITION 19,4:PRINT "**"
1290 FOR A=1 TO 10+RND(1)*20
1300 W=PEEK(764)
1310 POKE 764,255
1311 IF W=63 OR W=0 THEN GOTO 1315
1312 NEXT A
1314 GOTO 1210
1315 IF W=63 THEN P=P-M
1316 IF W=0 THEN P=P+M
1360 POSITION 0,6:PRINT " "
1365 POSITION 0,9:PRINT W$
1370 POSITION P,6:PRINT A$
1380 POSITION 0,7:PRINT " "
1390 POSITION P,7:PRINT B$
1400 POSITION 0,8:PRINT " "
1410 POSITION P,8:PRINT C$
1430 POSITION P,9:PRINT D$
1440 IF P=1 OR P=19 THEN 1460
1450 GOTO 1210
1460 POSITION 16,2
1470 IF P=1 THEN PRINT "VINCE A"
1480 IF P=19 THEN PRINT "VINCE L"
1485 POSITION 19,4:PRINT " "
1490 POSITION 5,19
1500 PRINT "Premere RETURN per giocare"
1510 POSITION 1,22
1520 INPUT R$
1530 RUN

```


ROULETTE

Il programma realizza una **Roulette** completa di sonoro. La richiesta di un numero avviene premendo il tasto N.

```
1000 REM * * *      ROULETTE      * * *
1010 REM -----
1020 REM -          di Gaetano Marano -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 23
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 712,0:POKE 708,12:POKE 709,54:POKE 710,202
1100 POKE 82,0
1110 REM -----
1120 PI=3.14159
1130 DIM P(37),I(37),N$(37),F(1)
1150 C=1:L=2:H=2:W=0
1155 RESTORE
1160 FOR Q=0 TO 353 STEP 9.729
1170 K=PI/180*Q
1180 X=68*COS(K)+78
1190 Y=44*SIN(K)+45
1200 READ I,N$
1210 I(C)=I:P(C)=VAL(N$):C=C+1
1220 IF I=2 OR I=3 THEN GOSUB 1380
1230 X=X-1
1250 N=ASC(N$(1,1)):N=N-48
1260 GOSUB 1680
1270 X=X+4
1280 N=ASC(N$(2,2)):N=N-48
1290 GOSUB 1680
1300 NEXT Q
1320 L=10:H=8
1330 R=37+INT(RND(0)*75)
1340 F=PEEK(764)
1350 POKE 764,255
1360 IF F=35 THEN GOSUB 1430
1370 GOTO 1330
1380 FOR K=Y-(H/2) TO Y+H*2+(H/2)
1390 COLOR 1
1392 IF I=1 THEN COLOR 0
1395 IF I=3 THEN COLOR 3
1400 PLOT X+W-(L/2),K:DRAWTO X+W+L*2+(L/2),K
1410 NEXT K
1420 RETURN
1430 V=0
1440 FOR F=0 TO R*9.729 STEP 9.729
1450 V=V+1:IF V=38 THEN V=1
1455 COLOR 0
1460 PLOT A,B
1470 SOUND 0,0,0,0
1500 K=PI/180*F
```

```

1510 A=INT(60*COS(K)+80)
1520 B=INT(37*SIN(K)+47)
1525 COLOR 1
1530 PLOT A,B
1540 SOUND 0,50,10,5
1580 NEXT F
1585 SOUND 0,0,0,0
1590 X=70:Y=40:W=2
1592 I=I(V)
1600 N$=STR$(P(V))
1610 GOSUB 1380
1611 IF LEN(N$)=1 THEN N=ASC(N$(1,1))-48:X=X+13:GOTO 1660
1620 N=ASC(N$(1,1)):N=N-48
1630 GOSUB 1680
1640 X=X+13
1650 N=ASC(N$(2,2)):N=N-48
1660 GOSUB 1680
1670 RETURN
1680 IF N=-16 THEN RETURN
1685 COLOR 2
1686 IF I=2 OR I=3 THEN COLOR 0
1690 IF N=0 OR N=2 OR N=3 OR N=5 OR N=6 OR N=7 OR N=8 OR N=9 THEN PLOT X,Y:DRAW
0 X+L,Y
1700 IF N=2 OR N=3 OR N=4 OR N=5 OR N=6 OR N=8 OR N=9 THEN PLOT X,Y+H:DRAWTO X+L
,Y+H
1710 IF N=0 OR N=2 OR N=3 OR N=5 OR N=6 OR N=8 OR N=9 THEN PLOT X,Y+H+H:DRAWTO X
+L,Y+H+H
1720 IF N=0 OR N=1 OR N=2 OR N=3 OR N=4 OR N=7 OR N=8 OR N=9 THEN PLOT X+L,Y:DRA
WTO X+L,Y+H
1730 IF N=0 OR N=1 OR N=3 OR N=4 OR N=5 OR N=6 OR N=7 OR N=8 OR N=9 THEN PLOT X+
L,Y+H:DRAWTO X+L,Y+H+H
1740 IF N=0 OR N=2 OR N=6 OR N=8 THEN PLOT X,Y+H:DRAWTO X,Y+H+H
1750 IF N=0 OR N=4 OR N=5 OR N=6 OR N=8 OR N=9 THEN PLOT X,Y:DRAWTO X,Y+H
1760 RETURN
2000 DATA 1,5,2,24,1,16,2,33,1,1,2
2010 DATA 20,1,14,2,31,1,9,2,22,1,18
2020 DATA 2,29,1,7,2,28,1,12,2,35
2030 DATA 1,3,2,26,3,0,1,32,2,15,1
2040 DATA 19,2,4,1,21,2,2,1,25,2,17
2050 DATA 1,34,2,6,1,27,2,13,1,36
2060 DATA 2,11,1,30,2,8,1,23,2,10

```

FLIPPER GIAPPONESE

In questo **Flipper Giapponese**, trenta palline cadono dall'alto attraverso una serie di chiodini, finendo in modo casuale su dei punteggi diversi che vengono automaticamente sommati al totale.

```
1000 REM * * FLIPPER GIAPPONESE * *
1010 REM -----
1020 REM - di Gaetano Marano -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1090 POKE 709,10:POKE 710,128:POKE 712,208
1100 POKE 752,1:POKE 82,0
1110 REM -----
1120 DIM X$(1),R$(1):X$=CHR$(20)
1160 POSITION 13,2:PRINT " "
1170 GOSUB 2000
1180 POSITION 7,16:PRINT "9 2 1 " 1 2 9"
1190 POSITION 7,17:PRINT "9 5 0 2 1 " 1 2 0 5 9"
1200 POSITION 7,18:PRINT "9 0 0 5 2 6 5 5 6 2 5 0 0 9"
1210 POSITION 1,22:PRINT "Premere RETURN"
1220 POSITION 16,22:INPUT R$
1235 POSITION 0,22:PRINT " "
1240 POSITION 18,22:PRINT " PUNTEGGIO":POSITION 36,22:PRINT " "
1250 S=0
1260 FOR A=1 TO 30
1270 L=2:C=20
1280 POSITION 4,4:PRINT A;" "
1290 POSITION 20,2:PRINT X$:FOR T=1 TO 150:NEXT T:POSITION 20,2:PRINT " ":POSITI
ON 20,2:PRINT X$:POSITION 20,2:PRINT ""
1300 FOR K=1 TO 13
1310 H=RND(1)
1320 IF H<0,5 THEN H=-1
1330 IF H>0,5 THEN H=1
1340 POSITION C,L:PRINT " "
1350 L=L+1
1360 POSITION C+H,L:PRINT X$
1370 SOUND 0,50,10,5
1380 C=C+H
1385 SOUND 0,0,0,0
1390 NEXT K
1400 IF C=19 OR C=21 THEN S=S+5
1410 IF C=17 OR C=23 THEN S=S+6
1420 IF C=15 OR C=25 THEN S=S+12
1430 IF C=13 OR C=27 THEN S=S+25
1440 IF C=11 OR C=29 THEN S=S+50
1450 IF C=9 OR C=31 THEN S=S+250
1460 IF C=7 OR C=33 THEN S=S+999
1470 POSITION 33,22:PRINT S
```

```

1480 FOR Q=1 TO 50:NEXT Q:POSITION C,L:PRINT " "
1490 NEXT A
1500 GOTO 1210
2000 DIM A$(30)
2010 A$=" . . . . . "
2020 B=18:C=3
2030 FOR A=0 TO 10 STEP 2
2040 POSITION B,C:PRINT A$(13-A,18+A)
2050 B=B-1:C=C+1
2060 POSITION B,C:PRINT A$(13-A,20+A)
2070 B=B-1:C=C+1
2080 NEXT A
2090 POSITION B,C:PRINT A$
2100 RETURN

```

ROMBOSPIRALE

Rombospirale è una animazione che disegna e cancella alternativamente e in continuazione, una spirale a forma di rombo formata da caratteri e simboli scelti casualmente.

```
1000 REM * *          ROMBOSPIRALE          * *
1010 REM -----
1020 REM -      di Gaetano Marano      -
1030 REM -----
1040 REM (c) Gruppo Editoriale Jackson
1050 REM -----
1060 GRAPHICS 0
1070 SOUND 0,0,0,0:SOUND 1,0,0,0
1075 SOUND 2,0,0,0:SOUND 3,0,0,0
1080 POKE 710,0:POKE 712,0:POKE 709,10
1100 POKE 82,0:POKE 752,1
1110 REM -----
1120 C=1
1130 GOSUB 1380
1140 FOR P=1 TO 11
1150 POSITION P+9,12-P:PRINT CHR$(X)
1160 NEXT P
1170 FOR N=1 TO 11 STEP 2
1180 GOSUB 1380
1190 FOR P=1 TO 12-N
1200 POSITION 19+P,-1+P+N:PRINT CHR$(X)
1210 NEXT P
1220 GOSUB 1380
1230 FOR P=1 TO 12-N
1240 POSITION 32-P-N,10+P:PRINT CHR$(X)
1250 NEXT P
1260 GOSUB 1380
1270 FOR P=1 TO 11-N
1280 POSITION 21-P,23-P-N:PRINT CHR$(X)
1290 NEXT P
1300 GOSUB 1380
1310 FOR P=1 TO 11-N
1320 POSITION 9+P+N,13-P:PRINT CHR$(X)
1330 NEXT P
1340 NEXT N
1350 C=C*(-1)
1360 X=32
1370 GOTO 1130
1380 T=INT(RND(1)*251)
1381 IF T=27 OR T=28 OR T=29 OR T=30 OR T=31 THEN GOTO 1380
1382 IF T=32 OR T=125 OR T=126 OR T=127 OR T=155 THEN GOTO 1380
1383 IF T=156 OR T=157 OR T=158 OR T=159 OR T=160 THEN GOTO 1380
1388 IF C=1 THEN X=T
1390 IF C=-1 THEN COLOR 1
1400 RETURN
```


I programmi di questo libro per l'Atari 130XE sono di tipo molto vario e vanno dai programmi di grafica, alle utility, ai giochi.

Tra gli altri vi sono dei programmi per disegnare, per programmare il generatore di suoni a quattro voci, per trasformare il computer in un piano, per giocare alla roulette o al tiro alla fune, ed altri ancora. Tutti i programmi utilizzano le interessanti caratteristiche del 130XE quali la grafica ad alta risoluzione, il generatore di suoni, ecc.

Dello stesso Autore:

- Programmi per Commodore 128
- Programmi per C16
- 30 programmi per MSX
- Programmi per Sinclair QL
- Macintosh 120 idee per il nuovo computer Apple
- Epson HX20 II Basic e 46 programmi
- 77 programmi per Spectrum
- 66 programmi per ZX81 e ZX80 + Hardware
- Gli amplificatori di Norton LM3900 & LM359

